

UNITED STATES DEPARTMENT OF INTERIOR

GEOLOGICAL SURVEY

Coal-seismic, desk-top computer programs in BASIC;
part 2: enter, compute, display, edit, and store
results of downhole, inhole, and crosshole investi-
gations

by

W. P. Hasbrouck

Open-File Report 80-669

1980

This report is preliminary and has not been edited
or reviewed for conformity to U.S. Geological
Survey standards

Coal-seismic, desk-top computer programs in BASIC; part 2:
enter, compute, display, edit, and store results of downhole,
in角度, and crosshole investigations

by

W. P. Hasbrouck

ABSTRACT

Processing of geophysical data taken with the U.S. Geological Survey's coal-seismic system is done with a desk-top, stand-alone computer. Programs for this computer are written in an extended BASIC language specially augmented for acceptance by the Tektronix 4051 Graphic System ^{1/}. This report presents computer programs used to enter, compute, display, edit, and store results of downhole, inhole, and crosshole seismic investigations.

INTRODUCTION

The purpose of this report is to make available four computer programs used to enter, compute, display, edit, and store results of downhole, inhole, and crosshole (hole-to-hole) seismic investigations. These data processing procedures were developed as part of the U.S. Geological Survey's coal-seismic system. All computer programs are written in an extended BASIC language developed by Tektronics, Inc. for use with their 4051 Graphic System. Three pieces of Tektronix computing equipment are required by the programs: a 4051 Graphic System with a 32,000-byte memory, a 4662 Interactive Digital Plotter, and a 4924 Digital Cartridge Tape Drive. A Tektronix 4631 Hard Copy unit can be used to make a copy of the screen display, if desired.

This report is divided into four sections, one for each program presented. The first three sections treat the downhole, inhole and crosshole survey

^{1/} Use of brand names in this report is for descriptive purposes only and does not constitute endorsement of the U.S. Geological Survey.

programs respectively; the fourth describes the program used to edit information stored on data tapes from all three types of surveys. The sections on the survey programs begin with a description of how the surveys are conducted and computed. This is followed by a series of figures that are copies of the displays produced on the CRT screen showing step-by-step examples of how data are entered and illustrating the results obtained. Each section ends with a presentation of the program.

One does not have to acquire extensive knowledge about the operation of the 4051 Graphic System before running the programs of this report. To get started, you'll need to know how to perform the following:

1. Enter a program from the keyboard,
2. Insert a magnetic-tape cartridge,
3. Transfer a program from memory to tape and back again,
4. Store data on tape, and
5. Enter information from the keyboard.

These tasks are well documented in the computer's operator's manuals.

All programs are self-prompting; that is, you don't have to remember what to do next--the program will tell you. In tracing through the sample problems you will notice that the programs will cause questions to be printed across the screen and that these questions will terminate with a flashing question mark. The computer then waits for you to supply an answer. Replies that you will need to enter from the keyboard are given in the examples--to make these responses easier to see, they are enclosed in boxes on the figures.

The sample problems use synthetic data obtained from a group of modeling programs (not included in this report). Model parameters of these forward calculations are listed in table 1. The idealized geologic section examined

Table 1 Model parameters

Layer Number	Layer Thickness (meters)	Layer Velocity (m/msec)
1	3.0	0.30
2	7.0	0.40
3	2.0	0.50
4	6.0	0.60
5	2.0	0.90
6	18.0	1.70
7	20.0	2.00
8	5.0	3.00
9	12.0	2.50
10	1.0	1.80
11	14.0	2.60
12	3.0	1.80
13	2.0	2.00
14	4.0	1.80
15	1.0	2.00
16	1.0	2.20
17	1.0	2.40
18	7.0	2.60
19	7.0	2.60
20	9.0	3.30
21	15.0	3.60

by the three velocity survey methods of this report contains a 1-m-thick coal seam (top at 75 m) and a 7-m-thick coal seam (top at 90 m) cut by a 2-m-thick parting. The floor under the lower coal shows a transitional increase in velocity indicative of the effect of an underclay. Depth of the LVL (low velocity layer) is 20 m, and total depth of the borehole is 140 m.

After the hole information and observed data are entered from the keyboard, they are stored on a magnetic tape inserted in the 4924 Digital Cartridge Tape Drive. Although it is not necessary to transfer observed data to a data tape (the program can proceed using keyboard-entered data only), it is recommended that this transfer be made for the following reasons:

1. The data tape can be readily edited should an error have occurred on keyboard input,
2. Because choice of final plotting scales and labeling is best made after

a trial plot, it is faster to read data from a file than have to re-enter them each time from the keyboard, and

3. It is good practice to save hard-won field data.

The programs of this report are listed with a printer that cannot print an underscore. This deficiency presents a problem in the display of control characters. Five display control characters are used in the program: G (rings bell), J (moves cursor down on line), K (moves cursor up one line), L (erases screen and moves cursor to Home position), and _ (moves cursor to the left margin and down one line). In the print of the program listing, these control characters are shown as G_, J_, K_, L_, and __ respectively.

THE DOWNHOLE SURVEY PROGRAM

Downhole surveys are used in coal-seismic studies to determine the average, RMS (root-mean-square), and interval velocities within the coal measures. In general, because the thickness of the coal seams is smaller than the thickness of oil-producing zones, the spacing between detectors in coal studies is less than the spacing between detectors in oil exploration. Also, in coal studies, the sampling interval of the seismograph system is usually smaller: 0.05 msec in coal work as contrasted to 1 msec or more in petroleum exploration.

Figure 1 shows the field setup for conducting a downhole velocity survey with a special multidetector cable. This cable is constructed such that either the velocity along the cable is lower than the velocity of the surrounding rocks, or the detectors are acoustically isolated from each other. Cables of this kind are called "velocity cables". In the sample problem, 11 downhole detectors and 1 surface detector, called the "reference seismometer", are used. The program is not restricted to 11 downhole detectors, but because of memory limitation of the computer, a maximum of 99 detector positions can be handled (9 cable positions with an 11-detector cable).




Figure 1. Field setup for conducting a velocity-cable downhole survey. Note that the top detector at the first cable position (detector 1,1) and the bottom detector at the second cable position (detector 2,11) are at the same depth in the test hole.

The downhole survey program has provisions for a survey run with the use of a single wall-lock seismometer. To select this branch of the program, an "N" would be entered in response to the question:

WERE DATA OBTAINED WITH USE OF A VELOCITY CABLE? (Y OR N).

Velocities are computed using first arrival times that have been adjusted for reference-time deviation and common-location time differences and then corrected to approximate vertical time. The approximation involves using the cosine of the angle whose tangent equals the source-to-test hole distance (also called the "hole offset") divided by the distance from the source point to the detector. Datum elevation is taken as the elevation of the first source position, and datum time is taken with reference to arrival times from the first source excitation--the first shot.

No unique solution exists to the problem of determining the velocity-layer structure from a series of first arrival times of seismic waves initiated at a single seismic source point and detected at a series of downhole positions. Any algorithm is thus an approximation. The results computed, displayed, and tabulated by this program must be considered only at first approximations.

Let us now work through a sample problem illustrating the use of the downhole survey program. Unless otherwise stated, in all the programs of this report, all distances, elevations, and depths are in meters (m); all arrival times are in milliseconds (msec); and all velocities are in meters per millisecond (m/msec).

Figure 2 is a copy of the first page of the screen display produced by the program after you have entered the information enclosed in the boxes.

Figures 3, 4, and 5 show the screen display after you have entered detector depths, arrival times to detectors at these depths, reference-seismometers arrival time, uphole times, and shotpoint (or source point) depths

YOU HAVE SELECTED PROGRAM TO ENTER, DISPLAY, COMPUTE
AND STORE DOWNHOLE VELOCITY SURVEY DATA

DO YOU WANT TO COMPUTE AND PLOT THOSE DATA STORED
ON AN DOWNHOLE-SURVEY DATA STORAGE TAPE? (Y OR N) ☐ N

Area name (20 characters,max) = Watkins test area
Hole designation (7 characters,max) = 79-101A
Date on which survey was made = Feb.11,1980

Latitude of test hole = 39 48 45.6
Longitude of test hole = 103 23 12.3
County and State (23 characters,max) = Adams County, Colorado

Elevation of test hole = 1650
Elevation of SP hole = 1650.2

Offset of test hole from SP = 5

Type of in-hole detector used (16 characters,max) Piezoelectric
Type of seismic source (12 characters,max) Kinepak, 1/3#

WERE DATA OBTAINED WITH USE OF A VELOCITY CABLE? (Y OR N) ☐ Y

Number of inhole cable positions = 3
Number of detectors on cable = 11

Figure 2. Copy of the screen display showing first page of information required by the downhole survey program. The replies to be entered from the keyboard in order to trace through the sample problems have been emphasized by enclosing the needed responses in boxes.

DETECTOR NUMBER, DEPTHS, ARRIVAL TIMES, REFERENCE AND UPHOLE TIMES

NUMBER DEPTH ARRIVAL TIME

1,1	90	67.8
1,2	94	70
1,3	98	72.1
1,4	102	74
1,5	106	75.7
1,6	110	77.1
1,7	114	78.5
1,8	118	79.8
1,9	122	81
1,10	126	82.1
1,11	130	83.2

Reference time for cable position 1 = 80.4

Uphole time for cable position 1 = 10.4

Shotpoint depth at cable position 1 = 3

Figure 3. Copy of screen display showing second page of information required by the downhole survey program. Data to be supplied here are those obtained when the velocity cable is in its first or lowest position. See fig. 1. Data to be entered are enclosed in boxes.

DETECTOR NUMBER, DEPTHS, ARRIVAL TIMES, REFERENCE AND UPHOLE TIMES

NUMBER DEPTH ARRIVAL TIME

2,1	50	51.7
2,2	54	53.6
2,3	58	55.6
2,4	62	56.9
2,5	66	58.4
2,6	70	60
2,7	74	61.7
2,8	78	63.4
2,9	82	64.9
2,10	86	66.4
2,11	90	68

Reference time for cable position 2 = 81.1

Uphole time for cable position 2 = 9.8

Shotpoint depth at cable position 2 = 2.9

Figure 4. Copy of screen display showing third page of information required by the downhole survey program. Although the top detector at one cable position and the bottom detector at the next cable position are at the same depth, note that their arrival times are not necessarily the same.

DETECTOR NUMBER, DEPTHS, ARRIVAL TIMES, REFERENCE AND UPHOLE TIMES

NUMBER DEPTH ARRIVAL TIME

3,1	10	23
3,2	14	28.7
3,3	18	34.7
3,4	22	37.4
3,5	26	39.5
3,6	30	41.6
3,7	34	44
3,8	38	46.2
3,9	42	48.2
3,10	46	50.2
3,11	50	52.1

Reference time for cable position 3 = 81.7

Uphole time for cable position 3 = 9.6

Shotpoint depth at cable position 3 = 2.8

Figure 5. Copy of screen display showing fourth page of information required by the downhole survey program. Here data observed when the velocity cable is in its third position are to be entered.

when the velocity cable was in its first, second, and third position respectively. In field practice, we read the uphole time from the CRT display on a single-trace seismograph, and then record this time and the shotpoint depth in a field notebook. Detector and reference-seismometer arrival times are obtained from the output of a 12-channel, signal-enhancement seismograph system.

Figure 6 is a copy of the fifth page of the sample problem screen display. Here we have selected to store the keyboard-entered information on a data tape, file 25. Then we have chosen to see those data. Upon receipt of a "Y" answer to the third question on figure 6, the program reads these data from the data tape and then prints them as shown on this figure and its continuation on figure 7. The arrow (lower part of fig. 6) points to an arrival-time entry that is obviously wrong. Thus, in answer to the question:

DO YOU WANT TO CHANGE ANY ARRIVAL TIMES? (Y OR N),
a "Y" would be entered, whereupon the display shown in figure 8 would be produced. Note that in entering the detector number, an ordered pair is required: in this case 1, 1. If seismometer 5 on cable 2 was the one at which the correction of arrival time was to be made, then you would enter a 2 followed by a 5 to the question at the top of figure 8.

Line 9 of figure 8 shows that we have chosen not to plot results on the 4622 (the Interactive Digital Plotter); rather, we want to see the results displayed on the screen. The plot limits and the tickmark intervals and their labels that are to be used on this plot are indicated in the boxes.

Figure 9 is a copy of the screen display showing plots of results obtained with the downhole survey program using the information entered in figures 2 through 8. This display can be thought of as a working copy.

Figure 10 is a copy of the screen display showing a tabulation of results of the sample downhole survey. In the calculations, the position of the

DO YOU WANT TO STORE OBSERVED DATA ON A TAPE? (Y OR N) ☒ Y

INSERT DOWNHOLE SURVEY DATA TAPE WITHIN 4924

WITHIN WHICH FILE ARE DATA TO BE STORED?

DO YOU WANT TO SEE TAPE OR ENTERED DATA? (Y OR N) ☒ Y

HEADER INFORMATION FROM FILE 25

NUMBER OF CABLE POSITIONS (N1) = 3
NUMBER OF SEIS/CABLE (N2) = 11
AREA (A\$) = Watkins test area
HOLE DESIGNATION (U\$) = 79-101A
COUNTY AND STATE (C\$) = Adams County, Colorado
DATE DATA TAKEN (D\$) = Feb. 11, 1980
LATITUDE (L\$) = 39 48 45.6 AND LONGITUDE (O\$) = 103 23 12.3
DETECTOR TYPE (R\$) = Piezoelectric AND SOURCE TYPE (S\$) = Kinepak, 1/3#
ELEV OF TEST HOLE (E1) = 1650; ELEV OF SP HOLE (E2) = 1650.2
SP TO TEST HOLE DIST (O1) = 5

CABLE NO.	REFERENCE TIME	SHOT DEPTH	UPHOLE TIME
J	R1(J)	S1(J)	T1(J)
1	80.40	3.0	10.4
2	81.10	2.9	9.8
3	81.70	2.8	9.6

FOR CABLE POSITION 1:		
SEIS NO.	ARRIVAL TIME	SEIS DEPTH
K	T5(1,K)	Z1(1,K)
1	6708.00	90.0
2	70.00	94.0
3	72.10	98.0
4	74.00	102.0
5	75.70	106.0



Figure 6. Copy of screen display showing the fifth page of information required by the downhole survey program. If an "N" is entered in response to either the first or third questions, then the program will skip ahead and produce the questions shown on figure 8. In this example, data were stored in file 25, and then those data were recalled and printed. Arrow points to incorrect arrival time.

FOR CABLE POSITION 2:		SEIS DEPTH
SEIS NO.	ARRIVAL TIME T5(2,K)	Z1(2,K)
6	77.10	110.0
7	78.50	114.0
8	79.80	118.0
9	81.00	122.0
10	82.10	126.0
11	83.20	130.0
1	51.70	50.0
2	53.60	54.0
3	55.60	58.0
4	56.90	62.0
5	58.40	66.0
6	60.00	70.0
7	61.70	74.0
8	63.40	78.0
9	64.90	82.0
10	66.40	86.0
11	68.00	90.0

FOR CABLE POSITION 3:		SEIS DEPTH
SEIS NO.	ARRIVAL TIME T5(3,K)	Z1(3,K)
1	23.00	10.0
2	28.70	14.0
3	34.70	18.0
4	37.40	22.0
5	39.50	26.0
6	41.60	30.0
7	44.00	34.0
8	46.20	38.0
9	48.20	42.0
10	50.20	46.0
11	52.10	50.0

Figure 7. Copy of screen display showing sixth page of information produced by the downhole survey program. This is a continuation on the listing began on figure 6.

MANUAL ENTRY OF VALUES FOR ADJUSTING ARRIVAL TIMES

NO. OF DETECTOR WHOSE ARRIVAL TIME IS TO BE ADJUSTED =

For detector 1,1 at a depth of 90,
old arrival time = 6708; new arrival time is to =

DO YOU WANT TO ADJUST OTHER ARRIVALS? (Y OR N)

DO YOU WANT TO STORE ADJUSTED VALUES? (Y OR N)

INSERT DOWNHOLE SURVEY DATA TAPE WITHIN 4924

WITHIN WHICH FILE ARE DATA TO BE STORED?

DO YOU WANT TO PLOT DATA ON 4662? (Y OR N)

ENTER PLOT-LIMIT VALUES

Min. depth of detector on plot =
Max. depth of detector on plot =

Min. arrival time on plot =
Max. arrival time on plot =

Min. velocity on plot =
Max. velocity on plot =

Ref. time deviation on plot =

TICKMARK INTERVALS ON PLOT

Depth tickmark interval =
Arrival-time tickmark interval =
Velocity tickmark interval =

Figure 8. Copy of screen display showing seventh page of information produced by the downhole survey program.

Here the program has brought in the section of the program for correcting entered arrival times. Next the choice is given as to whether results are to be plotted on the 4662 plotter or on the screen. Finally, plot limits and label intervals are called for.

DOWNHOLE SURVEY, HOLE:79-101A AREA:Watkins test area DATE:Feb.11,1980
 LAT: 39 48 45.6; LONG: 103 23 12.3; Adams County, Colorado FILE NO: 25
 HOLE ELEV=1650.0 SP ELEV=1650.2 AVG SP DEPTH= 2.9 AVG UH TIME= 9.9
 DETECTOR:Piezoelectric SEISMIC SOURCE:Kinopak,1/3# HOLE OFFSET= 5.0




Figure 9. Copy of screen display showing plot of results obtained with the downhole survey program using information entered in figures 2 through 8. Observed times are those read directly from the seismograms. Deviations from mean reference time (REF DEV) are plotted on the center panel. Interval (INT) velocities are shown by the stepped straight lines; average (AVG) velocities are plotted with an X symbol.

DOWNHOLE SURVEY, HOLE:79-101A AREA:Watkins test area DATE:Feb.11,1980
 LAT: 39 48 45.6; LONG: 103 23 12.3; Adams County, Colorado FILE NO: 25
 HOLE ELEV=1650.0 SP ELEV=1650.2 AVG SP DEPTH= 2.9 AVG UH TIME= 9.9
 DETECTOR:Piezoelectric SEISMIC SOURCE:Kinopak,1/3# HOLE OFFSET= 5.0

DETECTOR NO.	DEPTH	DATUM TIME	SURFACE V. TIME	VELOCITY IN M/MSEC		REMARKS
				INTERVAL	RMS. AVERAGE	
1	10.0	21.70	27.67	0.36	0.36	
2	14.0	27.40	34.87	0.41	0.40	
3	18.0	33.40	41.57	0.44	0.43	
4	22.0	36.10	44.78	1.25	0.49	
5	26.0	38.20	47.19	1.66	0.55	
6	30.0	40.30	49.48	1.74	0.61	
7	34.0	42.70	52.01	1.58	0.65	
8	38.0	44.90	54.30	1.75	0.70	
9	42.0	46.90	56.37	1.93	0.75	
10	46.0	48.90	58.42	1.95	0.79	
11	50.0	50.80	60.36	2.06	0.83	
12	54.0	52.70	62.30	2.07	0.87	
13	58.0	54.70	64.32	1.97	0.90	
14	62.0	56.00	65.65	1.12	0.94	
15	66.0	57.50	67.17	1.24	0.98	
16	70.0	59.10	68.78	1.28	1.02	
17	74.0	60.80	70.50	2.33	1.05	
18	78.0	62.50	72.21	2.34	1.08	
19	82.0	64.00	73.72	2.65	1.11	
20	86.0	65.50	75.23	2.65	1.14	
21	90.0	67.10	76.84	2.49	1.17	
22	94.0	69.50	79.24	1.66	1.19	
23	98.0	71.60	81.35	1.90	1.20	
24	102.0	73.50	83.25	2.10	1.23	
25	106.0	75.20	84.96	2.35	1.25	
26	110.0	76.60	86.36	2.85	1.27	
27	114.0	78.00	87.77	2.85	1.30	
28	118.0	79.30	89.07	3.07	1.32	
29	122.0	80.50	90.28	3.32	1.35	
30	126.0	81.60	91.38	3.62	1.38	
31	130.0	82.70	92.48	3.63	1.41	

Figure 10. Copy of screen display showing tabulation of datum time (observed time corrected to the first source position and corrected to remove observed-time differences at common-detector depths), vertical time from the surface, and velocities computed from vertical times.

first shot in the shothole is taken as the datum elevation, and all subsequent shot times are referenced to the first shot at the datum elevation. Three sets of velocities are listed in this tabulation: interval, RMS, and average. Each of these velocities has its own particular use in subsequent processing of seismic reflection data.

Shown in figure 11 is the screen display produced when a display on the 4662 plotter has been selected. Here instead of reentering the data, they have been called from file 25. Also we elected not to have the data printed. The lower half of the figure contains the plot limit and tickmark-interval entries.

Figure 12 is the result of entering the information of figure 11. This plot is generally acceptable for a publication or report. Information needed to write a figure caption for this figure, if it were to be part of a report, can be obtained from the 4-line label across the top of either the working plot (fig. 9) or the data tabulation (fig. 10). A full picture of the results of the downhole survey at hole 70-101A could be given in an engineering report by including a plot such as figure 12, a tabulation such as figure 10, and a listing of observed data such as shown in figures 6 and 7.

YOU HAVE SELECTED PROGRAM TO ENTER, DISPLAY, COMPUTE
AND STORE DOWNHOLE VELOCITY SURVEY DATA

DO YOU WANT TO COMPUTE AND PLOT THOSE DATA STORED
ON AN DOWNHOLE-SURVEY DATA STORAGE TAPE? (Y OR N)

INSERT UPHOLE SURVEY DATA TAPE WITHIN 4924

WITHIN WHICH FILE ARE DATA STORED?

DO YOU WANT TO SEE TAPE OR ENTERED DATA? (Y OR N)

DO YOU WANT TO PLOT DATA ON 4662? (Y OR N)

ENTER PLOT-LIMIT VALUES

Min. depth of detector on plot =
Max. depth of detector on plot =

Min. arrival time on plot =
Max. arrival time on plot =

Min. velocity on plot =
Max. velocity on plot =

TICKMARK INTERVALS ON PLOT

Depth tickmark interval =
Arrival-time tickmark interval =
Velocity tickmark interval =

Figure 11. Copy of screen display showing information required by the downhole survey program when a display on the 4662 plotter is to be made. Note that in this case, data are called from file 25.




Figure 12. Reduced copy of report-quality plot produced by the 4662 plotter after information of figure 11 was entered. Original dimensions of the outer border were 25.4 by 33.0 cm. Information needed to write a figure caption can be obtained from the 4-line label printed across the top of the screen display (fig. 9) or the data tabulation (fig. 10).

Listing of the program for downhole-survey data

```

100 PRINT "LL YOU HAVE SELECTED PROGRAM TO ENTER, DISPLAY, COMPUTE"
110 PRINT "      AND STORE DOWNHOLE VELOCITY SURVEY DATA"
120 INIT
130 DIM A$(20), C$(23), D$(12), F$(1), G$(1), H$(13), I$(12), L$(10), N$(24)
140 DIM O$(11), R$(16), S$(12), U$(7)
150 DATA 0, 25, 1, 32, "AVG SP DEPTH=", "AVG UH TIME="
160 READ B0, N3, P, H$, I$
170 N$="DETECTOR DEPTH IN METERS"
180 PRINT "___DO YOU WANT TO COMPUTE AND PLOT THOSE DATA STORED"
190 PRINT "ON AN DOWNHOLE-SURVEY DATA STORAGE TAPE? (Y OR N) ";
200 INPUT G$
210 IF G$="N" THEN 390
220 PRINT "G_G_G___INSERT UPHOLE SURVEY DATA TAPE WITHIN 4924"
230 PRINT "___WITHIN WHICH FILE ARE DATA STORED? ";
240 INPUT F1
250 FIND @2:F1
260 READ @2:N1, N2, N3
270 IF N3<=2 THEN 300
280 PRINT "G_G_G___ERROR---DATA FOR DOWNHOLE SURVEY NOT SELECTED"
290 GO TO 180
300 IF N3=2 THEN 320
310 GO TO 350
320 N1=1
330 DIM R1(N2), R2(N2), R3(N1, N2), S1(N2), T1(N2), T5(N1, N2), Z1(N1, N2)
340 GO TO 360
350 DIM R1(N1), R2(N1), R3(N1, N2), S1(N1), T1(N1), T5(N1, N2), Z1(N1, N2)
360 FIND @2:F1
370 READ @2:N1, N2, N3, A$, C$, D$, L$, O$, R$, S$, U$, E1, E2, O1, R1, S1, T1, T5, Z1
380 GO TO 420
390 F1=0
400 GOSUB 880
410 GOSUB 1250
420 PRINT "___DO YOU WANT TO SEE TAPE OR ENTERED DATA? (Y OR N) ";
430 INPUT G$
440 IF G$="N" THEN 470
450 GOSUB 6810
460 GOSUB 7320
470 Z8=S1(1)+E1-E2
480 IF N3=2 THEN 520
490 GOSUB 3850
500 GOSUB 3950
510 GO TO 540
520 N1=1
530 GOSUB 6590
540 PRINT "___DO YOU WANT TO PLOT DATA ON 4662? (Y OR N) ";
550 INPUT G$
560 IF G$="N" THEN 630
570 RESTORE 580
580 DATA 1, 1, 8, 3, 0, 5, 10, 95, 12, 122, 0, 5
590 READ P, K3, K4, B, B1, B2, B4, B5, C
600 B9=B2-B1
610 GOSUB 1560
620 GO TO 710
630 RESTORE 640
640 DATA 1, 792, 2, 816, 9, 88, 9, 56, 73, 129, 57, 72
650 READ K1, K2, B1, B2, B4, B5, B7, B8, C2, C3
660 B9=B8-B7
670 C4=C3-C2
680 B6=B5-B4
690 GOSUB 1560
700 GOSUB 7550
710 GOSUB 1830
720 GOSUB 2510
730 GOSUB 3280
740 GOSUB 2870
750 IF P=1 THEN 830

```

```

760 GOSUB 3160
770 GOSUB 3540
780 GOSUB 3710
790 GOSUB 4320
800 GOSUB 4600
810 GOSUB 4930
820 GO TO 860
830 GOSUB 5840
840 GOSUB 4320
850 GOSUB 6010
860 PRINT "G_G_G___PROGRAM COMPLETED"
870 END
880 REM ** SUB:ENTER SURVEY DESCRIPTION DATA
890 PRINT "___Area name (20 characters,max) = ";
900 INPUT A$
910 PRINT "Hole designation (7 characters,max) = ";
920 INPUT U$
930 PRINT "Date on which survey was made = ";
940 INPUT D$
950 PRINT "___Latitude of test hole = ";
960 INPUT L$
970 PRINT "Longitude of test hole = ";
980 INPUT O$
990 PRINT "County and State (23 characters,max) = ";
1000 INPUT C$
1010 PRINT "___Elevation of test hole = ";
1020 INPUT E1
1030 PRINT "Elevation of SP hole = ";
1040 INPUT E2
1050 PRINT "___Offset of test hole from SP = ";
1060 INPUT O1
1070 PRINT "___Type of in-hole detector used (16 characters,max) ";
1080 INPUT R$
1090 PRINT "Type of seismic source (12 characters,max) ";
1100 INPUT S$
1110 PRI "___WERE DATA OBTAINED WITH USE OF A VELOCITY CABLE? (Y OR N) ";
1120 INPUT G$
1130 IF G$="Y" THEN 1170
1140 GOSUB 6330
1150 GOSUB 1450
1160 GO TO 420
1170 PRINT "___Number of inhole cable positions = ";
1180 INPUT N1
1190 PRINT "Number of detectors on cable = ";
1200 INPUT N2
1210 DIM R1(N1), R2(N1), R3(N1, N2), S1(N1), T1(N1), T5(N1, N2), Z1(N1, N2)
1220 MOVE 0,0
1230 PRINT
1240 RETURN
1250 REM ** SUB: INPUT DETECTOR DEPTHS AND ARRIVAL TIMES
1260 FOR J=1 TO N1
1270 PRINT "L_DETECTOR NUMBER, DEPTHS, ARRIVAL TIMES, ";
1280 PRINT "REFERENCE AND UPHOLE TIMES__"
1290 PRINT " NUMBER"; " DEPTH"; " ARRIVAL TIME"
1300 FOR K=1 TO N2
1310 PRINT "___", J, ", ", K, " ";
1320 INPUT Z1(J, K)
1330 PRINT "K_";
1340 INPUT T5(J, K)
1350 NEXT K
1360 PRINT "___ Reference time for cable position "; J; " = ";
1370 INPUT R1(J)
1380 PRINT "___ Uphole time for cable position "; J; " = ";
1390 INPUT T1(J)
1400 PRINT "___ ShotPoint depth at cable position "; J; " = ";
1410 INPUT S1(J)
1420 MOVE 0,0
1430 PRINT
1440 NEXT J

```

```

1450 PRINT "___DO YOU WANT TO STORE OBSERVED DATA ON A TAPE?";
1460 PRINT " (Y OR N ) ";
1470 INPUT G$
1480 IF G$="N" THEN 1550
1490 PRINT "G_G_G___INSERT DOWNHOLE SURVEY DATA TAPE WITHIN 4924"
1500 PRINT "___WITHIN WHICH FILE ARE DATA TO BE STORED? ";
1510 INPUT F1
1520 FIND @2:F1
1530 WRITE @2:N1, N2, N3, A$, C$, D$, L$, O$, R$, S$, U$, E1, E2, O1, R1, S1, T1, T5, Z1
1540 PRINT @2, 2:
1550 RETURN
1560 REM ** SUB: ENTER PLOT LIMITS
1570 PRINT "___ENTER PLOT-LIMIT VALUES"
1580 PRINT "___  Min. depth of detector on plot = ";
1590 INPUT Z9
1600 PRINT "      Max. depth of detector on plot = ";
1610 INPUT Z0
1620 PRINT "___  Min. arrival time on plot = ";
1630 INPUT T9
1640 PRINT "      Max. arrival time on plot = ";
1650 INPUT T0
1660 PRINT "___  Min. velocity on plot = ";
1670 INPUT V9
1680 PRINT "      Max. velocity on plot = ";
1690 INPUT V0
1700 IF P=1 THEN 1730
1710 PRINT "___  Ref. time deviation on plot = ";
1720 INPUT R0
1730 PRINT "___TICKMARK INTERVALS ON PLOT"
1740 PRINT "___  Depth tickmark interval = ";
1750 INPUT M1
1760 PRINT "      Arrival-time tickmark interval = ";
1770 INPUT M2
1780 PRINT "      Velocity tickmark interval = ";
1790 INPUT M3
1800 MOVE 0,0
1810 PRINT
1820 RETURN
1830 REM ** SUB: PRINT HEADING, PLOT BORDERS, AND LABEL AXES
1840 B3=B2-B1
1850 B6=B5-B4
1860 IF P=32 THEN 1920
1870 MOVE @1:0,0
1880 RDRAW @1:0,100
1890 RDRAW @1:130,0
1900 RDRAW @1:0,-100
1910 RDRAW @1:-130,0
1920 MOVE @P:B4,B2
1930 RDRAW @P:B6,0
1940 RDRAW @P:0,-B3
1950 RDRAW @P:-B6,0
1960 RDRAW @P:0,B3
1970 IF P=1 THEN 2200
1980 MOVE C2,B2
1990 RDRAW C4,0
2000 RDRAW 0,-B3
2010 RDRAW -C4,0
2020 RDRAW 0,B3
2030 MOVE B7,B2
2040 RDRAW B9,0
2050 RDRAW 0,-B3
2060 RDRAW -B9,0
2070 RDRAW 0,B3
2080 MOVE 0,(B1+B2+(LEN(N$)-1.5)*K2)/2
2090 FOR I=1 TO LEN(N$)
2100 F$=SEG(N$, I, 1)
2110 PRINT F$
2120 NEXT I
2130 MOVE (B4+B5)/2-10.5*K1, B1-6

```

```

2140 PRINT "OBSERVED TIME IN MSEC", "K_"
2150 MOVE C2+K1, B1-6
2160 PRINT "REF DEV", "K_"
2170 MOVE (B7+B8-24*K1)/2, B1-6
2180 PRINT "INT & AVG VEL. IN M/MSEC", "K_"
2190 RETURN
2200 MOVE @1:8, 15.7
2210 PRINT @1, 25:90
2220 PRINT @1, 17:K3, K4
2230 PRINT @1: "VERTICAL TIME FROM SURFACE IN MSEC"
2240 MOVE @1:7, 89.6
2250 FOR J=1 TO 5
2260 RMOVE @1:0, -2
2270 GOSUB 5040
2280 NEXT J
2290 MOVE @1:128, 2
2300 PRINT @1: "INTERVAL"
2310 MOVE @1:127, 18
2320 RDRAW @1:0, 8
2330 MOVE @1:128, 29
2340 PRINT @1: "AND AVERAGE"
2350 MOVE @1:127, 49
2360 FOR J=1 TO 5
2370 RMOVE @1:0, 2
2380 GOSUB 5120
2390 NEXT J
2400 MOVE @1:128, 62
2410 PRINT @1: "VELOCITIES IN M/MSEC"
2420 MOVE @1:88, 5.3
2430 PRINT @1, 25:180
2440 PRINT @1:N$
2450 PRINT @1, 25:90
2460 PRINT @1, 17:3, 4, 5
2470 MOVE @1:4, 2.4
2480 PRINT @1: "DOWNHOLE SURVEY AT HOLE "; U$
2490 PRINT @1, 17:K3, K4
2500 RETURN
2510 REM ** SUB: PLOT AND LABEL DEPTH TICKMARKS
2520 IF P=32 THEN 2550
2530 C1=B6/(Z0-Z9)
2540 GO TO 2560
2550 C1=B3/(Z0-Z9)
2560 H1=M1*(INT(Z9/M1)+1)
2570 G1=(H1-Z9)*C1
2580 IF P=32 THEN 2610
2590 GOSUB 5200
2600 RETURN
2610 MOVE B4, B2
2620 RMOVE 0, -G1
2630 GOSUB 2710
2640 FOR I=1 TO INT((Z0-Z9)/M1)
2650 RMOVE 0, -M1*C1
2660 H1=H1+M1
2670 IF H1=>Z0 THEN 2700
2680 GOSUB 2710
2690 NEXT I
2700 RETURN
2710 REM ** SUB: TICKMARK PLOTTING
2720 RDRAW 0, 2, 0
2730 RMOVE B6-0, 4, 0
2740 RDRAW 0, 2, 0
2750 RMOVE 1, 0
2760 RDRAW 0, 2, 0
2770 RMOVE C4-0, 4, 0
2780 RDRAW 0, 2, 0
2790 RMOVE 1, 0
2800 RDRAW 0, 2, 0
2810 RMOVE B9-0, 4, 0
2820 RDRAW 0, 2, 0

```



```

2830 RMOVE -B6-B9-C4-7.4,-K2/2
2840 PRINT USING "3D":H1
2850 RMOVE 5.4,K2/2
2860 RETURN
2870 REM ** SUB: PLOT AND LABEL ARRIVAL TIME TICKMARKS
2880 IF P=32 THEN 2910
2890 C1=B3/(T0-T9)
2900 GO TO 2920
2910 C1=B6/(T0-T9)
2920 H1=M2*(INT(T9/M2)+1)
2930 G1=(H1-T9)*C1
2940 IF P=32 THEN 2970
2950 GOSUB 5430
2960 RETURN
2970 MOVE B4,B1
2980 RMOVE G1,0
2990 GOSUB 3010
3000 GO TO 3090
3010 REM ** SUB: TICKMARK PLOTTING
3020 RDRAW 0,0.2
3030 RMOVE 0,B3-0.4
3040 RDRAW 0,0.2
3050 RMOVE -1.7*K1,-B3-K2
3060 PRINT @P: USING "3D":H1
3070 RMOVE 1.7*K1,K2
3080 RETURN
3090 FOR I=1 TO INT((T0-T9)/M2)
3100 H1=H1+M2
3110 IF H1=>T0 THEN 3150
3120 RMOVE M2*C1,0
3130 GOSUB 3010
3140 NEXT I
3150 RETURN
3160 REM ** SUB: PLOT AND LABEL REF TIME DEVIATION TICKMARKS
3170 MOVE (C2+C3)/2,B1
3180 RDRAW 0,0.2
3190 RMOVE 0,B3-0.4
3200 RDRAW 0,0.2
3210 MOVE C2-1.3*K1,B1-K2
3220 PRINT USING "2D":-R0
3230 RMOVE K1+C4/2,0
3240 PRINT "0"
3250 RMOVE C4/2,0
3260 PRINT USING "D":R0
3270 RETURN
3280 REM ** SUB: PLOT AND LABEL VELOCITY TICKMARKS
3290 C1=B9/(V0-V9)
3300 H1=M3*(INT(V9/M3)+1)
3310 G1=(H1-V9)*C1
3320 IF P=32 THEN 3350
3330 GOSUB 5640
3340 RETURN
3350 MOVE B7,B1
3360 RMOVE G1,0
3370 GOSUB 3390
3380 GO TO 3470
3390 REM ** SUB: TICKMARK PLOTTING
3400 RDRAW 0,0.2
3410 RMOVE 0,B3-0.4
3420 RDRAW 0,0.2
3430 RMOVE -1.3*K1,-B3-K2
3440 PRINT USING "D.D":H1
3450 RMOVE 1.3*K1,K2
3460 RETURN
3470 FOR I=1 TO INT((V0-V9)/M3)
3480 H1=H1+M3
3490 IF H1=>V0-0.01 THEN 3530
3500 RMOVE M3*C1,0
3510 GOSUB 3390

```

```

3520 NEXT I
3530 RETURN
3540 REM ** SUB: PLOT ARRIVAL TIME VS. DEPTH
3550 WINDOW 0, T0-T9, 0, Z0-Z9
3560 VIEWPORT B4, B5, B1, B2
3570 C=B0*(T0-T9)/B6
3580 B=B0*(Z0-Z3)/B3
3590 DIM T2(N1, N2), Z2(N1, N2)
3600 T2=T5-T9
3610 Z2=Z0-Z1
3620 FOR J=N1 TO 1 STEP -1
3630 MOVE T2(J, 1), Z2(J, 1)
3640 GOSUB 5040
3650 FOR K=2 TO N2
3660 RDRAW T2(J, K)-T2(J, K-1), Z2(J, K)-Z2(J, K-1)
3670 GOSUB 5040
3680 NEXT K
3690 NEXT J
3700 RETURN
3710 REM ** SUB: PLOT REFERENCE TIME DEVIATIONS VS DEPTH
3720 WINDOW -R0, R0, 0, Z0-Z9
3730 VIEWPORT C2, C3, B1, B2
3740 C=B0*(R0+R0)/C4
3750 FOR J=N1 TO 1 STEP -1
3760 MOVE R3(J, 1), Z2(J, 1)
3770 GOSUB 5120
3780 FOR K=2 TO N2
3790 RDRAW R3(J, K)-R3(J, K-1), Z2(J, K)-Z2(J, K-1)
3800 GOSUB 5120
3810 NEXT K
3820 NEXT J
3830 DELETE T2, Z2
3840 RETURN
3850 REM ** SUB: CONDITION DATA FOR PLOTTING AND COMPUTING
3860 R6=SUM(R1)/N1
3870 S0=SUM(S1)/N1
3880 U0=SUM(T1)/N1
3890 FOR J=1 TO N1
3900 FOR K=1 TO N2
3910 R3(J, K)=R1(J)-R6
3920 NEXT K
3930 NEXT J
3940 RETURN
3950 REM ** CORRECT CAVITY DELAY AND REF TIME DIFFERENCES
3960 IF N1=1 THEN 4050
3970 R2(1)=0
3980 FOR J=2 TO N1
3990 R2(J)=0.5*(R1(J)-R1(1)+T1(J)-T1(1))
4000 R1(J)=R1(1)-R1(J)+R2(J)
4010 FOR K=1 TO N2
4020 T5(J, K)=T5(J, K)+R1(J)-R2(J)
4030 NEXT K
4040 NEXT J
4050 K2=N1*N2-N1+1
4060 DELETE R1, R2
4070 DIM T8(K2), Z5(K2), Z3(K2)
4080 L=0
4090 FOR K=1 TO N2
4100 T8(K)=T5(N1, K)
4110 Z5(K)=Z1(N1, K)
4120 Z3(K)=Z5(K)-Z8
4130 NEXT K
4140 IF N1=1 THEN 4250
4150 FOR J=N1-1 TO 1 STEP -1
4160 C2=T5(J+1, N2)-T5(J, 1)
4170 L=L+1
4180 K1=(N2-1)*L
4190 FOR K=2 TO N2
4200 T8(K1+K)=T5(J, K)+C2

```

```

4210 Z5(K1+K)=Z1(J,K)
4220 Z3(K1+K)=Z5(K1+K)-Z8
4230 NEXT K
4240 NEXT J
4250 REM ** PUT ARRIVAL TIME TO APPROXIMATE VERTICAL TIME
4260 DIM T3(K2)
4270 FOR M=1 TO K2
4280 T3(M)=T8(M)*COS(ATN(01/Z3(M)))
4290 NEXT M
4300 M=M-2
4310 RETURN
4320 REM ** SUB: COMPUTE INTERVAL(V1), RMS(V2), AND AVERAGE(V3) VELOCITY
4330 M=M+1
4340 DELETE R3, T5, Z1
4350 DIM T4(M), V1(M), V2(M), V3(M), Z4(M)
4360 V1(1)=(Z3(1)+S1(1))/(T3(1)+T1(1))
4370 V2(1)=V1(1)
4380 V3(1)=V1(1)
4390 T6=Z5(1)/V1(1)-T3(1)
4400 T3=T3+T6
4410 T4(1)=T3(1)
4420 Z4(1)=Z5(1)
4430 W=V1(1)*V1(1)*T3(1)
4440 X=Z4(1)
4450 Y=T3(1)
4460 FOR I=2 TO M
4470 T4(I)=T3(I)-T3(I-1)
4480 IF T4(I)=0 THEN 4500
4490 GO TO 4510
4500 T4(I)=-1
4510 Z4(I)=Z3(I)-Z3(I-1)
4520 V1(I)=Z4(I)/T4(I)
4530 W=W+V1(I)*V1(I)*T4(I)
4540 X=X+Z4(I)
4550 Y=Y+T4(I)
4560 V2(I)=SQR(W/Y)
4570 V3(I)=X/Y
4580 NEXT I
4590 RETURN
4600 REM ** SUB: PLOT VELOCITIES
4610 IF P=32 THEN 4640
4620 GOSUB 6010
4630 RETURN
4640 WINDOW 0, V0-V9, 0, Z0-Z9
4650 VIEWPORT B7, B8, B1, B2
4660 C=B0*(V0-V9)/B9
4670 V1=V1-V9
4680 V3=V3-V9
4690 Z5=Z0-Z5
4700 REM ** PLOT INTERVAL VELOCITIES ON CRT
4710 MOVE V1(1), Z0-Z9
4720 FOR K=1 TO 5
4730 RMOVE 0, -(Z0-Z9-Z5(1))/10
4740 RDRAW 0, -(Z0-Z9-Z5(1))/10
4750 NEXT K
4760 FOR K=2 TO M
4770 RDRAW V1(K)-V1(K-1), 0
4780 RDRAW 0, -Z4(K)
4790 NEXT K
4800 REM ** PLOT AVERAGE VELOCITIES ON CRT
4810 MOVE V3(1), Z5(1)
4820 GOSUB 5120
4830 FOR K=2 TO M
4840 RDRAW V3(K)-V3(K-1), Z5(K)-Z5(K-1)
4850 GOSUB 5120
4860 NEXT K
4870 Z5=Z0-Z5
4880 MOVE 0, 0
4890 PRINT "-----"

```

```

4900 V1=V1+V9
4910 V3=V3+V9
4920 RETURN
4930 REM ** SUB: TABULATE RESULTS
4940 GOSUB 7550
4950 PRINT "-- DETECTOR      DATUM      SURFACE      VELOCITY";
4960 PRINT " IN M/MSEC      REMARKS"
4970 PRINT "NO. DEPTH      TIME      V. TIME ";
4980 PRINT " INTERVAL RMS. AVERAGE"
4990 IMAGE 2D, X, 4D, D, 3X, 3D, 2D, 3X, 3D, 2D, 5X, 2D, 2D, 3X, 2D, 2D, 3X, 2D, 2D
5000 FOR I=1 TO M
5010 PRINT USING 4990: I, Z5(I), T8(I), T3(I), V1(I), V2(I), V3(I)
5020 NEXT I
5030 RETURN
5040 REM ** SUB: CROSS SYMBOL
5050 RDRAW @P:-C, 0
5060 RDRAW @P:2*C, 0
5070 RDRAW @P:-C, 0
5080 RDRAW @P:0, B
5090 RDRAW @P:0, -2*B
5100 RDRAW @P:0, B
5110 RETURN
5120 REM ** SUB: X SYMBOL
5130 RDRAW @P:-C, B
5140 RDRAW @P:2*C, -2*B
5150 RDRAW @P:-C, B
5160 RDRAW @P:-C, -B
5170 RDRAW @P:2*C, 2*B
5180 RDRAW @P:-C, -B
5190 RETURN
5200 REM ** SUB: DEPTH TICKMARKS AND VALUES, 4662 PLOT
5210 MOVE @1:12, 10
5220 K5=1
5230 K6=6.5
5240 K7=0.4
5250 RMOVE @1:G1, 0
5260 GOSUB 5340
5270 FOR I=1 TO INT((Z0-Z9)/M1)
5280 RMOVE @1:M1*C1, 0
5290 H1=H1+M1
5300 IF H1>Z0 THEN 5330
5310 GOSUB 5340
5320 NEXT I
5330 RETURN
5340 REM ** SUB: TICKMARK PLOTTING
5350 RDRAW @1:0, K7
5360 RMOVE @1:0, B3-2*K7
5370 RDRAW @1:0, K7
5380 RMOVE @1:0, -B3
5390 RMOVE @1:K5, -K6
5400 PRINT @1: USING "3D":H1
5410 RMOVE @1:-K5, K6
5420 RETURN
5430 REM ** SUB: TIME TICKMARKS AND VALUES, 4662 PLOT
5440 MOVE @1:12, 10
5450 K5=1
5460 K6=2.5
5470 RMOVE @1:0, G1
5480 GOSUB 5570
5490 FOR I=1 TO INT((T0-T9)/M2)
5500 RMOVE @1:0, M2*C1
5510 H1=H1+M2
5520 IF H1>T0 THEN 5560
5530 GOSUB 5570
5540 NEXT I
5550 PRINT @1, 7:
5560 RETURN
5570 REM ** SUB: TICKMARK PLOTTING
5580 RDRAW @1:K7, 0

```

```

5590 RDRAW @1:-K7,0
5600 RMOVE @1:-K5,-K6
5610 PRINT @1: USING "3D":H1
5620 RMOVE @1:K5,K6
5630 RETURN
5640 REM ** SUB: VELOCITY TICKMARKS AND VALUES, 4662 PLOT
5650 MOVE @1:122,10
5660 K5=3
5670 K6=-2.5
5680 RMOVE @1:0,G1
5690 GOSUB 5770
5700 FOR I=1 TO INT((V0-V9)/M3)
5710 H1=H1+M3
5720 IF H1>V0-0.01 THEN 5760
5730 RMOVE @1:0,M3*C1
5740 GOSUB 5770
5750 NEXT I
5760 RETURN
5770 REM ** SUB: TICKMARK PLOTTING
5780 RDRAW @1:-K7,0
5790 RDRAW @1:K7,0
5800 RMOVE @1:K5,K6
5810 PRINT @1: USING "D.D":H1
5820 RMOVE @1:-K5,-K6
5830 RETURN
5840 REM ** SUB: PLOT VERTICAL TIME FROM SURFACE VS. DEPTH, 4662 PLOT
5850 WINDOW 0,20-Z9,0,T0-T9
5860 VIEWPORT B4,B5,B1,B2
5870 DIM T7(M+1),Z7(M+1)
5880 B0=0.5
5890 C=B0*(Z0-Z9)/B6
5900 B=B0*(T0-T9)/B3
5910 T7=T3-T9
5920 Z7=Z5-Z9
5930 MOVE @1:Z7(1),T7(1)
5940 GOSUB 5040
5950 FOR I=2 TO M+1
5960 RDRAW @1:Z7(I)-Z7(I-1),T7(I)-T7(I-1)
5970 GOSUB 5040
5980 NEXT I
5990 DELETE T7,Z7
6000 RETURN
6010 REM ** SUB: PLOT INTERVAL AND AVERAGE VELOCITIES, 4662 PLOT
6020 WINDOW 0,20-Z9,0,V0-V9
6030 VIEWPORT B4,B5,B1,B2
6040 B0=0.5
6050 B=B0*(V0-V9)/B3
6060 V1=V1-V9
6070 V3=V3-V9
6080 REM **PLOT INTERVAL VELOCITIES
6090 MOVE @1:0,V1(1)
6100 FOR K=1 TO 5
6110 RMOVE @1:(Z5(1)-Z9)/10,0
6120 RDRAW @1:(Z5(1)-Z9)/10,0
6130 NEXT K
6140 FOR I=2 TO M
6150 RDRAW @1:0,V1(I)-V1(I-1)
6160 RDRAW @1:Z4(I),0
6170 NEXT I
6180 REM ** PLOT AVEPAGE VELOCITIES
6190 MOVE @1:Z5(1)-Z9,V3(1)
6200 B=0.8*B
6210 C=0.8*C
6220 GOSUB 5120
6230 FOR I=2 TO M
6240 RDRAW @1:Z5(I)-Z5(I-1),V3(I)-V3(I-1)
6250 GOSUB 5120
6260 NEXT I
6270 V1=V1+V9

```

```

6280 V3=V3+V9
6290 WINDOW 0,150,0,100
6300 VIEWPORT 0,150,0,100
6310 MOVE @1:150,100
6320 RETURN
6330 REM ** SUB: DATA ENTRY WHEN VELOCITY CABLE NOT USED
6340 PRINT "L_NOTE: WHEN DATA FROM A SINGLE-DOWNHOLE-SEISMOMETER SURVEY"
6350 PRINT " ARE ENTERED, NO DUPLICATE-DEPTH ARRIVALS ARE ALLOWED."
6360 PRINT "___Number of downhole seismometer positions = ";
6370 INPUT N2
6380 N1=1 6385 N3=2
6390 DIM R1(N2), R2(N2), R3(N1, N2), S1(N2), T8(N2), T1(N2), T2(N1, N2), T3(N2)
6400 DIM T5(N1, N2), T7(N2), Z5(N2), Z1(N1, N2), Z2(N1, N2), Z3(N2), Z7(N2)
6410 PRINT "___ENTER: DETECTOR DEPTH, ARRIVAL TIME, REF. TIME, ";
6420 PRINT " UH TIME, AND SP DEPTH__"
6430 PRINT "N0. "; " DEPTH"; " ARRIVAL TIME"; " REF TIME"; " UH TIME";
6440 PRINT " SP DEPTH"
6450 FOR J=1 TO N2
6460 PRINT " "; J; " ";
6470 INPUT Z1(1, J)
6480 Z5(J)=Z1(1, J)
6490 PRINT "K_";
6500 INPUT T5(1, J)
6510 PRINT "K_";
6520 INPUT R1(J)
6530 PRINT "K_";
6540 INPUT T1(J)
6550 PRINT "K_";
6560 INPUT S1(J)
6570 NEXT J
6580 RETURN
6590 REM ** SUB: CONDITION DATA FOR COMPUTING AND PLOTTING
6600 R6=SUM(R1)/N2
6610 S0=SUM(S1)/N2
6620 U0=SUM(T1)/N2
6630 FOR K=1 TO N2
6640 R3(1, K)=R1(K)-R6
6650 NEXT K
6660 R2(1)=0
6670 FOR J=2 TO N2
6680 R2(J)=0.5*(R1(J)-R1(1)+T1(J)-T1(1))
6690 R1(J)=R1(1)-R1(J)+R2(J)
6700 T5(1, J)=T5(1, J)+R1(J)-R2(J)
6710 NEXT J
6720 DIM T3(N2), T8(N2), Z3(N2), Z5(N2)
6730 FOR J=1 TO N2
6740 T8(J)=T5(1, J)
6750 Z5(J)=Z1(1, J)
6760 Z3(J)=Z5(J)-Z8
6770 T3(J)=T8(J)*COS(ATN(01/Z3(J)))
6780 NEXT J
6790 M=N2-1
6800 RETURN
6810 REM ** SUB: PRINT DATA ENTERED FROM TAPE OR KEYBOARD
6820 PRINT "___ HEADER INFORMATION FROM FILE "; F1
6830 IF N3=2 THEN 6870
6840 PRINT "NUMBER OF CABLE POSITIONS (N1) ="; N1
6850 PRINT "NUMBER OF SEIS/CABLE (N2) ="; N2
6860 GO TO 6880
6870 PRINT "NUMBER OF DETECTOR POSITIONS = "; N2
6880 PRINT "AREA (A$) = "; A$
6890 PRINT "HOLE DESIGNATION (U$) = "; U$
6900 PRINT "COUNTY AND STATE (C$) = "; C$
6910 PRINT "DATE DATA TAKEN (D$) = "; D$
6920 PRINT "LATITUDE (L$) = "; L$; " AND LONGITUDE (O$) = "; O$
6930 PRINT "DETECTOR TYPE (R$)= "; R$; " AND SOURCE TYPE (S$)= "; S$
6940 PRINT "ELEV OF TEST HOLE (E1) = "; E1; " ELEV OF SP HOLE (E2) = "; E2
6950 PRINT "SP TO TEST HOLE DIST (01) = "; 01
6960 IF N3=2 THEN 6980

```

```

6970 GO TO 7010
6980 DIM R1(N2), S1(N2), T1(N2), T5(1, N2), Z1(1, N2)
6990 N1=N2
7000 GO TO 7020
7010 DIM R1(N1), S1(N1), T1(N1), T5(N1, N2), Z1(N1, N2)
7020 IF F1=0 THEN 7050
7030 FIND @2:F1
7040 READ @2:N1, N2, N3, A$, C$, D$, L$, O$, R$, S$, U$, E1, E2, O1, R1, S1, T1, T5, Z1
7050 IF N3=1 THEN 7100
7060 N1=N2
7070 PRINT "__SEIS. NO. "; " REFERENCE TIME "; " SHOT DEPTH "; " UP";
7080 PRINT "HOLE TIME"
7090 GO TO 7130
7100 PRINT "__CABLE NO. "; " REFERENCE TIME "; " SHOT DEPTH "; " UP";
7110 PRINT "HOLE TIME"
7120 IMAGE 2X, 2D, 9X, 3D, 2D, 9X, 3D, D, 9X, 3D, D
7130 PRINT "      J              R1(J)              S1(J)              T1(J)"
7140 FOR J=1 TO N1
7150 PRINT USING 7120: J, R1(J), S1(J), T1(J)
7160 NEXT J
7170 PRINT
7180 J=1
7190 IF N3=2 THEN 7230
7200 FOR J=1 TO N1
7210 IF N3=2 THEN 7230
7220 PRINT "FOR CABLE POSITION "; J; ":"
7230 PRINT "SEIS NO. "; " ARRIVAL TIME"; " SEIS DEPTH"
7240 PRINT "      K              T5(J, K)              Z1(J, K)"
7250 FOR K=1 TO N2
7260 IMAGE 2X, 2D, 8X, 4D, 2D, 9X, 3D, D
7270 PRINT USING 7260: K, T5(J, K), Z1(J, K)
7280 NEXT K
7290 IF N3=2 THEN 7310
7300 NEXT J
7310 RETURN
7320 REM ** SUB: ADJUST OR CORRECT ARRIVAL TIME
7330 PRINT "__DO YOU WANT TO ADJUST (OR CORRECT) ARRIVAL TIMES?";
7340 PRINT " (Y OR N) ";
7350 INPUT G$
7360 IF G$="N" THEN 7540
7370 PRINT "LMANUAL ENTRY OF VALUES FOR ADJUSTING ARRIVAL TIMES"
7380 PRINT "__NO. OF DETECTOR WHOSE ARRIVAL TIME IS TO BE ADJUSTED = ";
7390 INPUT J
7400 PRINT "K_
7410 INPUT K
7420 PRINT "      For detector "; J; ", "; K; " at a depth of "; Z1(J, K); ", "
7430 PRINT "old arrival time = "; T5(J, K); "; new arrival time is to = ";
7440 INPUT T5(J, K)
7450 PRINT "__DO YOU WANT TO ADJUST OTHER ARRIVALS? (Y OR N) ";
7460 INPUT G$
7470 IF G$="N" THEN 7490
7480 GO TO 7380
7490 IF F1=0 THEN 7540
7500 PRINT "__DO YOU WANT TO STORE ADJUSTED VALUES? (Y OR N) ";
7510 INPUT G$
7520 IF G$="N" THEN 7540
7530 GOSUB 1490
7540 RETURN
7550 REM ** SUB: PRINT HEADING ON CRT
7560 WINDOW 0, 130, 0, 100
7570 VIEWPORT 0, 130, 0, 100
7580 IMAGE "LDOWNHOLE SURVEY, HOLE: ", 7A, " AREA: ", 20A, "DATE: ", 12A
7590 PRINT USING 7580: U$, A$, D$
7600 IMAGE "LAT: ", 10A, "; LONG: ", 11A, "; ", 23A, " FILE NO: ", 3D
7610 PRINT USING 7600: L$, O$, C$, F1
7620 IMAGE "HOLE ELEV=", 4D, D, 3X, "SP ELEV=", 4D, D, 3X, 13A, 2D, D, 3X, 12A, 2D, D
7630 PRINT USING 7620: E1, E2, H$, S0, I$, U0
7640 IMAGE "DETECTOR: ", 16A, " SEISMIC SOURCE: ", 12A, 2X, "HOLE OFFSET=", 2D, D
7650 PRINT USING 7640: R$, S$, O1
7660 RETURN

```

THE INHOLE SURVEY PROGRAM

An inhole survey, as we use the term, is one in which both the seismic source and detectors are positioned within the same borehole. This data-acquisition procedure is used to produce a more detailed examination of the velocity structure both within and bounding a coal-bed sequence. By placing the seismic source closer to the detectors and by using a less powerful source, it is possible to develop higher frequency arrivals at the detectors than if the source were located near the surface. Not only would the near-surface source have to be stronger, but also the down-going seismic wave would have to travel farther and do so through the more absorptive near-surface layers.

Figure 13 shows a typical field setup for conducting an inhole survey. In this example, the source is positioned 3.5 detector intervals beneath the lowest detector on the velocity cable. To avoid damaging the velocity cable, an explosive-type seismic source must be located below the velocity cable.

Interval velocities are computed by dividing the distance between a detector pair by the difference in first-arrival time to that detector pair; average velocities are computed by dividing the distance from the source to the detector by the first-arrival time to that detector. No correction for offset of the check-shot (surface) seismometer from the test hole is made because in most application this offset is considerably smaller than the source depth; thus, a cosine correction would be negligible.

In the sample problem that follows, the section from a depth of 68 to 108 m is examined using four inhole arrays. Separation between each of the 11 velocity-cable detectors is 1 m, and the bottom detector is 5 m above the source of each case. First-arrival times to the inhole detectors are entered to the nearest 0.01 msec, and check-shot times (the first-arrival time from the source to the surface seismometer) are entered to the nearest 0.1 msec.




Figure 13. Field setup for conducting a velocity-cable inhole survey. In this example, the seismic source is positioned 3.5 detector intervals beneath the number 12 detector on the velocity cable.

As in the previous sample problem, those quantities to be entered from the keyboard are enclosed in boxes.

Figure 14 is a copy of the screen display showing the first page of information required by the inhole survey program. The program requires that the RETURN key be struck after each entry and that the HOME-PAGE key be struck after the last entry on each page.

Figures 15 through 18 are copies of the screen display after the shot depth, check-shot time, detector depths, and arrival times at the detectors have been entered for each of the four cable positions.

In figure 19, the command has been given to store the entered data on file 26 and then recall and print these data. Figures 20 and 21 are copies of the screen display showing a continuation of the print of data within file 26. In this case, no entry errors are obvious. Thus in answer to the first question at the top of the figure 22, a "N" was entered.

Figure 22 is a copy of the screen display showing the responses to be given for plot limits, tickmark intervals, and choice of plot device.

Figure 23 is a copy of the screen display of the plots produced by the inhole survey program after entries as shown in figures 14 through 22 have been made. Note that the display of the interval velocities clearly indicates lower velocity material is present in the section being studied. There appear to be two thin coal beds (if, indeed, from experience in the area we had established that low velocity layers were most probably coal beds) near the top of the investigation section, and there appears to be either two thicker beds or perhaps one thick bed and a 2-m parting near the center of the detailed section. Also, the material beneath the lowest coal beds appears to show a gradual increase in velocity with depth.

YOU HAVE SELECTED PROGRAM TO ENTER, COMPUTE,
DISPLAY, AND STORE INHOLE SURVEY DATA

DO YOU WANT TO COMPUTE AND PLOT THOSE DATA STORED
ON AN INHOLE-SURVEY DATA TAPE? (Y OR N) **H**

Area name (20 characters,max) = **Watkins test area**
Hole designation (7 characters,max) = **79-101A**
Date on which survey was made = **Feb.13,1980**

Latitude of test hole = **39 48 45.6**
Longitude of test hole = **103 23 12.3**
County and State (23 characters,max) = **Adams County, Colorado**

Elevation of test hole = **1650**
Elevation of check-shot seismometer = **1650.2**
Offset of check-shot seismometer = **2**

Type of in-hole detector used (16 characters,max) **Piezoelectric**
Type of seismic source (12 characters,max) **Shotgun**

Number of cable positions = **4**
Number of detectors on cable = **11**

Figure 14. Copy of screen display showing first page of information required by the inhole survey program.
The replies to be entered from the keyboard in order to trace through the sample problem have been
emphasized by enclosing the needed responses in boxes.

DEPTH OF SHOT AT SHOT POSITION 1 = 113

DETECTOR DEPTH, ARRIVAL TIME, AND CHECK-SHOT TIME AT CABLE POSITION 1

TIME AT CHECK-SHOT SEIS = 87.5

NO. DEPTH ARRIVAL TIME

2	98	6
3	99	5.44
4	100	4.94
5	101	4.49
6	102	4.07
7	103	3.69
8	104	3.3
9	105	2.92
10	106	2.53
11	107	2.15
12	108	1.76

Figure 15. Copy of screen display showing the second page of information required by the inhole survey program. Data to be supplied here are those obtained when the inhole array is in its first or lowest position. See fig. 13. Data to be entered are enclosed in boxes.

DEPTH OF SHOT AT SHOT POSITION 2 = 103

DETECTOR DEPTH, ARRIVAL TIME, AND CHECK-SHOT TIME AT CABLE POSITION 2

TIME AT CHECK-SHOT SEIS = 83.8

NO. DEPTH ARRIVAL TIME

2	88	7.18
3	89	6.8
4	90	6.41
5	91	5.86
6	92	5.3
7	93	4.75
8	94	4.36
9	95	3.98
10	96	3.42
11	97	2.87
12	98	2.31

Figure 16. Copy of screen display showing third page of information required by the inhole survey program. Data to be supplied here are those obtained when the inhole array is in its second position. Note that the top detector at one cable position and the bottom detector at the next cable position are at the same depth.

DEPTH OF SHOT AT SHOT POSITION 3 = 93

DETECTOR DEPTH, ARRIVAL TIME, AND CHECK-SHOT TIME AT CABLE POSITION 3

TIME AT CHECK-SHOT SEIS = 79.1

NO. DEPTH ARRIVAL TIME

2	78	6.28
3	79	5.9
4	80	5.51
5	81	5.13
6	82	4.74
7	83	4.36
8	84	3.97
9	85	3.59
10	86	3.21
11	87	2.82
12	88	2.44

Figure 17. Copy of screen display showing fourth page of information required by the inhole survey program.
Data to be supplied here are those obtained when the inhole array is in its third position.

DEPTH OF SHOT AT SHOT POSITION 4 = 83

DETECTOR DEPTH, ARRIVAL TIME, AND CHECK-SHOT TIME AT CABLE POSITION 4

TIME AT CHECK-SHOT SEIS = 74.7

NO. DEPTH ARRIVAL TIME

2	68	6.05
3	69	5.65
4	70	5.25
5	71	4.85
6	72	4.45
7	73	4.05
8	74	3.65
9	75	3.25
10	76	2.69
11	77	2.31
12	78	1.82

Figure 18. Copy of screen display showing fifth page of information required by the inhole survey program.
Data to be supplied here are those obtained when the inhole array is in its last position.

DO YOU WANT TO STORE OBSERVED DATA ON A TAPE? <Y OR N> ☒ Y

INSERT INHOLE SURVEY DATA TAPE WITHIN 4924

WITHIN WHICH FILE ARE DATA TO BE STORED? ☒ 26

DO YOU WANT TO SEE TAPE OR ENTERED DATA? <Y OR N> ☒ Y

HEADER INFORMATION FROM FILE 26

NUMBER OF CABLE POSITIONS (N1) = 4

NUMBER OF SEIS/CABLE (N2) = 12

IDENTIFICATION NUMBER ON TYPE OF SURVEY (N3) = 3

N3=3 indicates an inhole survey was run

AREA (A\$) = Watkins test area

HOLE DESIGNATION (U\$) = 79-101A

COUNTY AND STATE (C\$) = Adams County, Colorado

DATE DATA TAKEN (D\$) = Feb. 13, 1980

LATITUDE (L\$) = 39 48 45.6 AND LONGITUDE (O\$) = 103 23 12.3

DETECTOR TYPE (R\$) = Piezoelectric AND SOURCE TYPE (S\$) = Shotgun

ELEV OF TEST HOLE (E1) = 1650; ELEV OF CHECK SEIS (E2) = 1650.2

CHECK SEIS OFFSET (O1) = 2

CABLE NO.	CHECK	SHOT TIME	SHOT DEPTH
J	T5(J,1)		91(J)
1	87.50		113.0
2	83.80		103.0
3	79.10		93.0
4	74.70		83.0

FOR CABLE POSITION 1:

SEIS NO.	ARRIVAL TIME	SEIS DEPTH
K	T5(1,K)	25(1,K)
1	87.50	-0.2
2	6.00	98.0

Figure 19. Copy of screen display showing sixth page of information required by the inhole survey program. In this example, data entered from the keyboard were stored in file 26, and then a print of the contents of that file was called by entering a "Y" to the third question.

3	5.44	99.0
4	4.94	100.0
5	4.49	101.0
6	4.07	102.0
7	3.69	103.0
8	3.30	104.0
9	2.92	105.0
10	2.53	106.0
11	2.15	107.0
12	1.76	108.0

FOR CABLE POSITION 2:		SEIS DEPTH
SEIS NO.	ARRIVAL TIME	25(2,K)
K	T5(2,K)	-0.2
1	83.80	88.0
2	7.18	89.0
3	6.80	90.0
4	6.41	91.0
5	5.86	92.0
6	5.30	93.0
7	4.75	94.0
8	4.36	95.0
9	3.98	96.0
10	3.42	97.0
11	2.87	98.0
12	2.31	

FOR CABLE POSITION 3:		SEIS DEPTH
SEIS NO.	ARRIVAL TIME	25(3,K)
K	T5(3,K)	-0.2
1	79.10	78.0
2	6.28	79.0
3	5.90	80.0
4	5.51	81.0
5	5.13	

Figure 20. Copy of the screen display showing a continuation of figure 19 of the listing of data from file 26.

6	4.74	82.0
7	4.36	83.0
8	3.97	84.0
9	3.59	85.0
10	3.21	86.0
11	2.82	87.0
12	2.44	88.0

FOR CABLE POSITION 4:		
SEIS NO.	ARRIVAL TIME	SEIS DEPTH
K	T5(4,K)	Z5(4,K)
1	74.70	-0.2
2	6.05	68.0
3	5.65	69.0
4	5.25	70.0
5	4.85	71.0
6	4.45	72.0
7	4.05	73.0
8	3.65	74.0
9	3.25	75.0
10	2.69	76.0
11	2.31	77.0
12	1.82	78.0

Figure 21. Copy of the screen display showing a continuation of figure 20 of the listing of data from file 26.

DO YOU WANT TO CHANGE ANY ARRIVAL TIMES? (Y OR N) ☐ N

ENTER PLOT-LIMIT VALUES

Min. depth of detector on plot = ☐ 66
Max. depth of detector on plot = ☐ 115

Min. arrival time on plot = ☐ 0
Max. arrival time on plot = ☐ 8

Min. velocity on plot = ☐ 1
Max. velocity on plot = ☐ 3

TICKMARK INTERVALS ON PLOT

Depth tickmark interval = ☐ 5
Arrival-time tickmark interval = ☐ 1
Velocity tickmark interval = ☐ .5

DO YOU WANT TO PLOT ON THE 4662? (Y OR N) ☐ N

Figure 22. Copy of screen display produced by the inhole survey program when it asked for plot limits, label intervals, and choice of plotter. An "N" answer to the last query causes the screen plot to be made.

INHOLE SURVEY, HOLE:79-101A AREA:Watkins test area DATE:Feb.13,1980
 LAT: 39 48 45.6; LONG: 103 23 12.3; Adams County, Colorado FILE NO:26
 HOLE ELEV=1650.0 CHECK SEIS. ELEV=1650.2 CHECK SEIS. OFFSET= 2.0
 DETECTOR:Piezoelectric SEISMIC SOURCE:Shotgun




Figure 23. Copy of screen display showing plot of results obtained with the inhole survey program using information entered in figures 14 through 22. The appearance of the interval velocity versus depth curve (over the depth interval from 76 to 90 m, for example) is jagged because in this example arrival times were entered only to the nearest 10 microsec.

Figures 24 and 25 are copies of the screen display showing tabulations of observed data and computed results for the inhole-survey sample problem. Now let us assume that prior mapping in this area has not revealed the existence of two thin coal beds near the depth of about 75 m. And let us further assume that we have gone back to the original data and determined that the arrival time at the 4,12 detector was in error. The display at the bottom of figure 25 shows how this correction in arrival time would be entered.

Figure 26 shows a copy of the screen display of the plots after the corrected value has been stored in file 26 and the program rerun, this time calling the data from a file rather than reentering them from the keyboard. Figure 27 is a copy of the last half of the data tabulation showing the corrected value. Note on figures 26 and 27 how a change in arrival time of only 0.1 msec can markedly change the results--the lower thin coal bed no longer appears to exist.

Figure 28 shows the information to be entered when a report-quality plot from the 4662 plotter is to be made. The result of entering the information on figure 28 is shown in figure 29. Even though the interval velocity plot is jagged, it nevertheless shows in far more detail than the downhole survey (fig. 12) the depths and thicknesses of probable coal beds. Also, a synthetic seismogram constructed with use of those velocity variation shown on figure 29 would have a better chance of looking like the observed seismograms of a high-resolution reflection survey than if the synthetic seismogram had been built upon the results obtained with a downhole survey.

INHOLE SURVEY, HOLE:79-101A AREA:Watkins test area DATE:Feb.13,1980
 LAT: 39 48 45.6; LONG: 103 23 12.3; Adams County, Colorado FILE NO:26
 HOLE ELEV=1650.0 CHECK SEIS. ELEV=1650.2 CHECK SEIS. OFFSET= 2.0
 DETECTOR:Piezoelectric SEISMIC SOURCE:Shotgun

DETECTOR NO.	SHOTPOINT DEPTH	SP-DETECTOR DIST.	TIME	VELOCITIES INTERVAL AVERAGE	REMARKS
1, 1	-0.2	113.0	87.50	1.29	CHECK SHOT
1, 2	98.0	113.0	6.00	1.79	
1, 3	99.0	113.0	5.44	2.00	
1, 4	100.0	113.0	4.94	2.22	
1, 5	101.0	113.0	4.49	2.38	
1, 6	102.0	113.0	4.07	2.63	
1, 7	103.0	113.0	3.69	2.71	
1, 8	104.0	113.0	3.30	2.73	
1, 9	105.0	113.0	2.92	2.74	
1, 10	106.0	113.0	2.53	2.77	
1, 11	107.0	113.0	2.15	2.79	
1, 12	108.0	113.0	1.76	2.84	
2, 1	-0.2	103.0	83.80	1.23	CHECK SHOT
2, 2	88.0	103.0	7.18	2.09	
2, 3	89.0	103.0	6.80	2.06	
2, 4	90.0	103.0	6.41	2.03	
2, 5	91.0	103.0	5.86	2.05	
2, 6	92.0	103.0	5.30	2.08	
2, 7	93.0	103.0	4.75	2.11	
2, 8	94.0	103.0	4.36	2.06	
2, 9	95.0	103.0	3.98	2.01	
2, 10	96.0	103.0	3.42	2.05	
2, 11	97.0	103.0	2.87	2.09	
2, 12	98.0	103.0	2.31	2.16	

Figure 24. Copy of screen display showing tabulation of observed data and computed results. Check shot time is the arrival time at the surface seismometer. See fig. 13.

	1	2	3	4	5	6	7	8	9	10	11	12	CHECK SHOT
5	1	2	3	4	5	6	7	8	9	10	11	12	1.18
5	1	2	3	4	5	6	7	8	9	10	11	12	2.39
5	1	2	3	4	5	6	7	8	9	10	11	12	2.37
5	1	2	3	4	5	6	7	8	9	10	11	12	2.36
5	1	2	3	4	5	6	7	8	9	10	11	12	2.34
5	1	2	3	4	5	6	7	8	9	10	11	12	2.32
5	1	2	3	4	5	6	7	8	9	10	11	12	2.29
5	1	2	3	4	5	6	7	8	9	10	11	12	2.27
5	1	2	3	4	5	6	7	8	9	10	11	12	2.23
5	1	2	3	4	5	6	7	8	9	10	11	12	2.18
5	1	2	3	4	5	6	7	8	9	10	11	12	2.13
5	1	2	3	4	5	6	7	8	9	10	11	12	2.05

	1	2	3	4	5	6	7	8	9	10	11	12	CHECK SHOT
4	1	2	3	4	5	6	7	8	9	10	11	12	1.11
4	1	2	3	4	5	6	7	8	9	10	11	12	2.48
4	1	2	3	4	5	6	7	8	9	10	11	12	2.48
4	1	2	3	4	5	6	7	8	9	10	11	12	2.48
4	1	2	3	4	5	6	7	8	9	10	11	12	2.47
4	1	2	3	4	5	6	7	8	9	10	11	12	2.47
4	1	2	3	4	5	6	7	8	9	10	11	12	2.47
4	1	2	3	4	5	6	7	8	9	10	11	12	2.46
4	1	2	3	4	5	6	7	8	9	10	11	12	2.60
4	1	2	3	4	5	6	7	8	9	10	11	12	2.60
4	1	2	3	4	5	6	7	8	9	10	11	12	2.75

DO YOU WANT TO CHANGE ANY ARRIVAL TIMES? (Y OR N) Y

ENTER VALUES TO CHANGE ARRIVAL TIMES

For detector number (ordered pair), see tabulated values

NO. OF DETECTOR WHOSE ARRIVAL TIME IS TO BE CORRECTED = 4,12

For detector 4,12 at a depth of 78,

old arrival time = 1.82; new arrival time is to = 1.92

Figure 25. Copy of screen display showing tabulation of observed data and computed results for the in-hole cable at its third and fourth positions. Also shown is the arrival-time change procedure.

INHOLE SURVEY, HOLE:79-101A AREA:Watkins test area DATE:Feb.13,1980
 LAT: 39 48 45.6; LONG: 103 23 12.3; Adams County, Colorado FILE NO:26
 HOLE ELEV=1650.0 CHECK SEIS. ELEV=1650.2 CHECK SEIS. OFFSET= 2.0
 DETECTOR:Piezoelectric SEISMIC SOURCE:Shotgun




Figure 26. Copy of screen display showing plot of results obtained with the inhole survey program after arrival-time correction was made. (See bottom of fig. 25.) Plot parameters are the same as on figure 22. Note the effect of an arrival-time change of only 0.1 msec on the interval velocity within the depth range of 77 and 78 m by comparing this plot to the one shown on figure 23.

CHECK SHOT

3, 1	-0.2	93.0	93.0	93.2	79.10	1.18	1.18
3, 2	78.0	93.0	93.0	15.0	6.28	2.63	2.39
3, 3	79.0	93.0	93.0	14.0	5.90	2.56	2.37
3, 4	80.0	93.0	93.0	13.0	5.51	2.63	2.36
3, 5	81.0	93.0	93.0	12.0	5.13	2.56	2.34
3, 6	82.0	93.0	93.0	11.0	4.74	2.63	2.32
3, 7	83.0	93.0	93.0	10.0	4.36	2.56	2.29
3, 8	84.0	93.0	93.0	9.0	3.97	2.63	2.27
3, 9	85.0	93.0	93.0	8.0	3.59	2.63	2.23
3, 10	86.0	93.0	93.0	7.0	3.21	2.56	2.18
3, 11	87.0	93.0	93.0	6.0	2.82	2.63	2.13
3, 12	88.0	93.0	93.0	5.0	2.44	2.05	2.05

CHECK SHOT

4, 1	-0.2	93.0	93.0	93.2	74.70	1.11	1.11
4, 2	68.0	93.0	93.0	15.0	6.05	2.50	2.48
4, 3	69.0	93.0	93.0	14.0	5.65	2.50	2.48
4, 4	70.0	93.0	93.0	13.0	5.25	2.50	2.48
4, 5	71.0	93.0	93.0	12.0	4.85	2.50	2.47
4, 6	72.0	93.0	93.0	11.0	4.45	2.50	2.47
4, 7	73.0	93.0	93.0	10.0	4.05	2.50	2.47
4, 8	74.0	93.0	93.0	9.0	3.65	2.50	2.47
4, 9	75.0	93.0	93.0	8.0	3.25	1.79	2.46
4, 10	76.0	93.0	93.0	7.0	2.69	2.63	2.60
4, 11	77.0	93.0	93.0	6.0	2.31	2.56	2.60
4, 12	78.0	93.0	93.0	5.0	1.92	2.60	2.60

DO YOU WANT TO CHANGE ANY ARRIVAL TIMES? (Y OR N) N

PROGRAM COMPLETED

Figure 27. Copy of screen display showing last half of tabulation of observed data and computed results after arrival-time correction was made. (See bottom of fig. 25.) First half of tabulation is the same as that shown on figure 24. Note change in velocities produced by arrival-time correction at detector 4,12.

YOU HAVE SELECTED PROGRAM TO ENTER, COMPUTE,
DISPLAY, AND STORE INHOLE SURVEY DATA

DO YOU WANT TO COMPUTE AND PLOT THOSE DATA STORED
ON AN INHOLE-SURVEY DATA TAPE? (Y OR N) ☒ Y ☐ N

INSERT INHOLE-SURVEY DATA TAPE WITHIN 4924

WITHIN WHICH FILE ARE DATA STORED?

DO YOU WANT TO SEE TAPE OR ENTERED DATA? (Y OR N) ☐ Y ☒ N

ENTER PLOT-LIMIT VALUES

Min. depth of detector on plot =
Max. depth of detector on plot =

Min. arrival time on plot =
Max. arrival time on plot =

Min. velocity on plot =
Max. velocity on plot =

TICKMARK INTERVALS ON PLOT

Depth tickmark interval =
Arrival-time tickmark interval =
Velocity tickmark interval =

DO YOU WANT TO PLOT ON THE 4662? (Y OR N) ☐ Y ☒ N

Figure 28. Copy of screen display showing information required by the inhole survey program when a display on the 4662 plotter is to be made. In this case, data are called from file 26.




Figure 29. Reduced copy of report-quality plot produced by the 4662 plotter after information of figure 11 was entered. Original dimensions of the outer border were 25.4 by 33.0 cm. Information needed to write a figure caption can be obtained either from the 4-line label printed across the top of the screen display plot (fig. 23) or the data tabulation (fig. 24).

Listing of program for inhole-survey data

```

100 PRINT "L_                                YOU HAVE SELECTED PROGRAM TO ENTER, COMPUTE, "
110 PRINT "                                DISPLAY, AND STORE INHOLE SURVEY DATA"
120 INIT
130 DIM A$(20), B$(31), C$(23), D$(12), F$(3), G$(1), H$(24), L$(10)
140 DIM N$(15), O$(11), R$(16), S$(12), U$(7), W$(13)
150 DATA 32, 1, "DEPTH IN METERS", "ARRIVAL TIME IN MILLISEC"
160 READ P, Q, N$, H$
170 PRINT "___DO YOU WANT TO COMPUTE AND PLOT THOSE DATA STORED"
180 PRINT "ON AN INHOLE-SURVEY DATA TAPE? (Y OR N) ";
190 INPUT G$
200 IF G$="N" THEN 330
210 PRINT "G_G_G___INSERT INHOLE-SURVEY DATA TAPE WITHIN 4924"
220 PRINT "___WITHIN WHICH FILE ARE DATA STORED? ";
230 INPUT F1
240 FIND @2:F1
250 READ @2:N1, N2, N3
260 IF N3=3 THEN 290
270 PRINT "G_G_G___ERROR--INHOLE SURVEY DATA NOT SELECTED"
280 GO TO 170
290 DIM R1(N1), S1(N1), T5(N1, N2), Z5(N1, N2)
300 FIND @2:F1
310 READ @2:N1, N2, N3, A$, C$, D$, L$, O$, R$, S$, U$, E1, E2, O1, R1, S1, S1, T5, Z5
320 GO TO 350
330 F1=0
340 GOSUB 590
350 PRINT "___DO YOU WANT TO SEE TAPE OR ENTERED DATA? (Y OR N) ";
360 INPUT G$
370 IF G$="N" THEN 400
380 GOSUB 4960
390 GOSUB 5320
400 GOSUB 1210
410 GOSUB 1430
420 GOSUB 1620
430 GOSUB 2200
440 GOSUB 2520
450 GOSUB 2810
460 DIM S2(N1)
470 FOR J=1 TO N1
480 S2(J)=Z5(J, 12)-Z5(J, 11)
490 NEXT J
500 GOSUB 3070
510 GOSUB 3370
520 GOSUB 3280
530 GOSUB 3490
540 IF P=1 THEN 570
550 GOSUB 3720
560 GOSUB 5320
570 PRINT "G_G_G___PROGRAM COMPLETED"
580 END
590 REM ** SUB:ENTER SURVEY DESCRIPTION DATA
600 PRINT "___Area name (20 characters,max) = ";
610 N3=3
620 INPUT A$
630 PRINT "Hole designation (7 characters,max) = ";
640 INPUT U$
650 PRINT "Date on which survey was made = ";
660 INPUT D$
670 PRINT "___Latitude of test hole = ";
680 INPUT L$
690 PRINT "Longitude of test hole = ";
700 INPUT O$
710 PRINT "County and State (23 characters,max) = ";
720 INPUT C$
730 PRINT "___Elevation of test hole = ";
740 INPUT E1
750 PRINT "Elevation of check-shot seismometer = ";

```

```

760 INPUT E2
770 PRINT "Offset of check-shot seismometer = ";
780 INPUT O1
790 PRINT "___Type of in-hole detector used (16 characters,max) ";
800 INPUT R$
810 PRINT "Type of seismic source (12 characters,max) ";
820 INPUT S$
830 PRINT "___Number of cable positions = ";
840 INPUT N1
850 PRINT "Number of detectors on cable = ";
860 INPUT N2
870 N2=N2+1
880 DIM S1(N1), T5(N1, N2), Z5(N1, N2)
890 MOVE 0,0
900 PRINT
910 REM ** INPUT DETECTOR DEPTHS AND ARRIVAL TIMES
920 FOR J=1 TO N1
930 Z5(J,1)=E1-E2
940 PRINT "___DEPTH OF SHOT AT SHOT POSITION "; J; " = ";
950 INPUT S1(J)
960 PRINT "___DETECTOR DEPTH, ARRIVAL TIME, AND CHECK-SHOT TIME";
970 PRINT " AT CABLE POSITION "; J
980 PRINT "___TIME AT CHECK-SHOT SEIS = ";
990 INPUT T5(J,1)
1000 PRINT "___ NO. "; " DEPTH"; " ARRIVAL TIME"
1010 FOR I=2 TO N2
1020 PRINT "___ "; I; " ";
1030 INPUT Z5(J, I)
1040 PRINT "K_ ";
1050 INPUT T5(J, I)
1060 NEXT I
1070 MOVE 0,0
1080 PRINT
1090 NEXT J
1100 PRINT "___DO YOU WANT TO STORE OBSERVED DATA ON A TAPE?";
1110 PRINT " (Y OR N ) ";
1120 INPUT G$
1130 IF G$="N" THEN 1210
1140 PRINT "G_G_G___INSERT INHOLE SURVEY DATA TAPE WITHIN 4924"
1150 PRINT "___WITHIN WHICH FILE ARE DATA TO BE STORED? ";
1160 INPUT F1
1170 FIND @2:F1
1180 WRITE @2:N1, N2, N3, A$, C$, D$, L$, O$, R$, S$, U$, E1, E2, O1, S1, S1, S1, T5, Z5
1190 PRINT @2, 2:
1200 RETURN
1210 REM ** SUB: ENTER PLOT LIMITS
1220 PRINT "___ENTER PLOT-LIMIT VALUES"
1230 PRINT "___ Min. depth of detector on plot = ";
1240 INPUT Z9
1250 PRINT " Max. depth of detector on plot = ";
1260 INPUT Z0
1270 PRINT "___ Min. arrival time on plot = ";
1280 INPUT T9
1290 PRINT " Max. arrival time on plot = ";
1300 INPUT T0
1310 PRINT "___ Min. velocity on plot = ";
1320 INPUT V9
1330 PRINT " Max. velocity on plot = ";
1340 INPUT V0
1350 PRINT "___TICKMARK INTERVALS ON PLOT"
1360 PRINT "___ Depth tickmark interval = ";
1370 INPUT M1
1380 PRINT " Arrival-time tickmark interval = ";
1390 INPUT M2
1400 PRINT " Velocity tickmark interval = ";
1410 INPUT M3
1420 RETURN
1430 REM ** SUB: PRINT SURVEY-INFORMATION DATA
1440 PRINT "___DO YOU WANT TO PLOT ON THE 4662? (Y OR N) ";

```

```

1450 INPUT G$
1460 IF G$="Y" THEN 1590
1470 MOVE 0,0
1480 PRINT
1490 IMAGE"L_INHOLE SURVEY, HOLE:",7A,2X,"AREA:",20A,X,"DATE:",12A
1500 PRINT USING 1490:U$,A$,D$
1510 IMAGE "LAT: ",10A,"; LONG: ",11A,"; ",23A," FILE NO:",2D
1520 PRINT USING 1510:L$,0$,C$,F1
1530 W$="CHECK SEIS. "
1540 IMAGE"HOLE ELEV=",4D,D,5X,11A," ELEV=",4D,D,5X,11A," OFFSET=",2D,D
1550 PRINT USING 1540:E1,W$,E2,W$,01
1560 IMAGE"DETECTOR:",16A,20X,"SEISMIC SOURCE:",12A
1570 PRINT USING 1560:R$,S$
1580 RETURN
1590 RESTORE 1600
1600 DATA 0.75,0.75,0.75,1,2,3,10,95,12,122
1610 READ C,B,B0,P,K3,K4,B1,B2,B4,B5
1620 RETURN
1630 REM ** SUB: PLOT BORDERS AND LABEL AXES
1640 WINDOW 0,130,0,100
1650 VIEWPORT 0,130,0,100
1660 IF P=1 THEN 1720
1670 RESTORE 1680
1680 DATA 0.5,1.792,2.816,9.88,9.69,70,129
1690 READ B0,K1,K2,B1,B2,B4,B5,B7,B8
1700 B9=B8-B7
1710 GO TO 1780
1720 MOVE @1:0,0
1730 RDRAW @1:0,100
1740 RDRAW @1:130,0
1750 RDRAW @1:0,-100
1760 RDRAW @1:-130,0
1770 B9=B2-B1
1780 B3=B2-B1
1790 B6=B5-B4
1800 MOVE @P:B4,B2
1810 RDRAW @P:B6,0
1820 RDRAW @P:0,-B3
1830 RDRAW @P:-B6,0
1840 RDRAW @P:0,B3
1850 IF P=1 THEN 2010
1860 MOVE B7,B2
1870 RDRAW B9,0
1880 RDRAW 0,-B3
1890 RDRAW -B9,0
1900 RDRAW 0,B3
1910 MOVE 0,(B1+B2+(LEN(N$)-1.5)*K2)/2
1920 FOR I=1 TO LEN(N$)
1930 F$=SEG(N$,I,1)
1940 PRINT F$
1950 NEXT I
1960 MOVE (B4+B5)/2-12*K1,B1-6
1970 PRINT H$,"K_"
1980 MOVE (B7+B8-29*K1)/2,B1-6
1990 PRINT "INTERVAL VELOCITIES IN M/MSEC","K_"
2000 RETURN
2010 MOVE @1:8,22
2020 PRINT @1,25:90
2030 PRINT @1:H$
2040 MOVE @1:7,84
2050 FOR J=1 TO 7
2060 RMOVE @1:0,-2
2070 GOSUB 3890
2080 NEXT J
2090 MOVE @1:128,20.5
2100 PRINT @1:"INTERVAL VELOCITY IN METERS/MILLISEC"
2110 MOVE @1:80,3
2120 PRINT @1,25:180
2130 PRINT @1:N$

```

```

2140 PRINT @1, 25:90
2150 MOVE @1:4, 7
2160 PRINT @1, 17:3, 4, 5
2170 PRINT @1: "INHOLE SURVEY AT HOLE "; U$
2180 PRINT @1, 17:K3, K4
2190 RETURN
2200 REM ** SUB: PLOT AND LABEL DEPTH TICKMARKS
2210 IF P=32 THEN 2240
2220 C1=B6/(Z0-Z9)
2230 GO TO 2250
2240 C1=B3/(Z0-Z9)
2250 H1=M1*(INT(Z9/M1)+1)
2260 G1=(H1-Z9)*C1
2270 IF P=32 THEN 2300
2280 GOSUB 3970
2290 RETURN
2300 MOVE B4, B2
2310 RMOVE 0, -G1
2320 GOSUB 2400
2330 FOR I=1 TO INT((Z0-Z9)/M1)
2340 RMOVE 0, -M1*C1
2350 H1=H1+M1
2360 IF H1=>Z0 THEN 2390
2370 GOSUB 2400
2380 NEXT I
2390 RETURN
2400 REM ** SUB: TICKMARK PLOTTING
2410 RDRAW 0, 2, 0
2420 RMOVE B6-0, 4, 0
2430 RDRAW 0, 2, 0
2440 RMOVE 1, 0
2450 RDRAW 0, 2, 0
2460 RMOVE B9-0, 4, 0
2470 RDRAW 0, 2, 0
2480 RMOVE -B6-B9-6, 4, -K2/2
2490 PRINT USING "3D":H1
2500 RMOVE 5, 4, K2/2
2510 RETURN
2520 REM ** SUB: PLOT AND LABEL ARRIVAL-TIME TICKMARKS
2530 IF P=32 THEN 2560
2540 C1=B3/(T0-T9)
2550 GO TO 2570
2560 C1=B6/(T0-T9)
2570 H1=M2*(INT(T9/M2)+1)
2580 G1=(H1-T9)*C1
2590 IF P=32 THEN 2620
2600 GOSUB 4200
2610 RETURN
2620 MOVE B4, B1
2630 RMOVE G1, 0
2640 GOSUB 2660
2650 GO TO 2740
2660 REM ** SUB: TICKMARK PLOTTING
2670 RDRAW 0, 0, 2
2680 RMOVE 0, B3-0, 4
2690 RDRAW 0, 0, 2
2700 RMOVE -1, 7*K1, -B3-K2
2710 PRINT @P: USING "3D":H1
2720 RMOVE 1, 7*K1, K2
2730 RETURN
2740 FOR I=1 TO INT((T0-T9)/M2)
2750 H1=H1+M2
2760 IF H1=>T0 THEN 2800
2770 RMOVE M2*C1, 0
2780 GOSUB 2660
2790 NEXT I
2800 RETURN
2810 REM ** SUB: PLOT AND LABEL VELOCITY TICKMARKS
2820 C1=B9/(V0-V9)

```

```

2830 H1=M3*(INT(V9/M3)+1)
2840 G1=(H1-V9)*C1
2850 IF P=32 THEN 2880
2860 GOSUB 4400
2870 RETURN
2880 MOVE B7,B1
2890 RMOVE G1,0
2900 GOSUB 2920
2910 GO TO 3000
2920 REM ** SUB: TICKMARK PLOTTING
2930 RDRAW 0,0.2
2940 RMOVE 0,B3-0.4
2950 RDRAW 0,0.2
2960 RMOVE -1.3*K1,-B3-K2
2970 PRINT USING "D.D":H1
2980 RMOVE 1.3*K1,K2
2990 RETURN
3000 FOR I=1 TO INT((V0-V9)/M3)
3010 H1=H1+M3
3020 IF H1>V0-0.01 THEN 3060
3030 RMOVE M3*C1,0
3040 GOSUB 2920
3050 NEXT I
3060 RETURN
3070 REM ** SUB: PLOT INHOLE ARRIVAL TIME VS. DEPTH
3080 DIM T2(N1,N2),Z2(N1,N2)
3090 IF P=32 THEN 3120
3100 GOSUB 4610
3110 RETURN
3120 WINDOW 0,T0-T9,0,Z0-Z9
3130 VIEWPORT B4,B5,B1,B2
3140 C=B0*(T0-T9)/B6
3150 B=B0*(Z0-Z9)/B3
3160 T2=T5-T9
3170 Z2=Z0-Z5
3180 FOR J=1 TO N1
3190 MOVE 0,Z0-S1(J)
3200 RDRAW T2(J,N2),Z2(J,N2)-Z0+S1(J)
3210 GOSUB 3890
3220 FOR K=N2 TO 3 STEP -1
3230 RDRAW T2(J,K-1)-T2(J,K),S2(J)
3240 GOSUB 3890
3250 NEXT K
3260 NEXT J
3270 RETURN
3280 REM ** SUB: COMPUTE INTERVAL VELOCITIES
3290 FOR J=1 TO N1
3300 V1(J,1)=V3(J,1)
3310 V1(J,N2)=V3(J,N2)
3320 FOR K=2 TO N2-1
3330 V1(J,K)=S2(J)/(T5(J,K)-T5(J,K+1))
3340 NEXT K
3350 NEXT J
3360 RETURN
3370 REM ** SUB: COMPUTE AVERAGE VELOCITIES
3380 DIM D3(N1,N2),V1(N1,N2),V2(N1,N2),V3(N1,N2)
3390 FOR J=1 TO N1
3400 V3(J,1)=S1(J)/T5(J,1)
3410 D3(J,1)=S1(J)+E2-E1
3420 Z5(J,1)=E1-E2
3430 FOR K=2 TO N2
3440 D3(J,K)=S1(J)-Z5(J,K)
3450 V3(J,K)=D3(J,K)/T5(J,K)
3460 NEXT K
3470 NEXT J
3480 RETURN
3490 REM ** SUB: PLOT INTERVAL VELOCITIES
3500 IF P=32 THEN 3530
3510 GOSUB 4790

```



```

3520 RETURN
3530 WINDOW 0, V0-V9, 0, Z0-Z9
3540 VIEWPORT B7, B8, B1, B2
3550 C=B0*(V0-V9)/B9
3560 V2=V1-V9
3570 Z2=Z0-Z5
3580 MOVE V1(N1, 2)-V9, Z2(N1, 2)
3590 FOR J=N1 TO 1 STEP -1
3600 DRAW V2(J, 2), Z2(J, 2)
3610 RDRAW 0, -S2(J)
3620 FOR K=2 TO N2-2
3630 RDRAW V2(J, K+1)-V2(J, K), 0
3640 RDRAW 0, -S2(J)
3650 NEXT K
3660 NEXT J
3670 RDRAW V2(1, N2)-V2(1, N2-1), 0
3680 RDRAW 0, Z5(1, N2)-S1(1)
3690 MOVE 0, 0
3700 PRINT "-----"
3710 RETURN
3720 REM ** SUB: TABULATE RESULTS
3730 GOSUB 1490
3740 PRINT "---- DETECTOR SHOTPOINT SP-DETECTOR ";
3750 PRINT "VELOCITIES REMARKS"
3760 PRINT " NO. DEPTH DEPTH DIST. TIME ";
3770 PRINT "INTERVAL AVERAGE--"
3780 IMAGE 2D, " ", 2D, 4D, D, 4X, 3D, D, 3X, 3D, D, X, 3D, 2D, 3X, 2D, 2D, 3X, 2D, 2D
3790 FOR J=1 TO N1
3800 FOR K=1 TO N2
3810 PRINT USING 3780: J, K, Z5(J, K), S1(J), D3(J, K), T5(J, K), V1(J, K), V3(J, K)
3820 IF K=2 THEN 3850
3830 PRINT "K_"
3840 PRINT "CHECK SHOT"
3850 NEXT K
3860 PRINT
3870 NEXT J
3880 RETURN
3890 REM ** SUB: CROSS SYMBOL
3900 RDRAW @P: -C, 0
3910 RDRAW @P: 2*C, 0
3920 RDRAW @P: -C, 0
3930 RDRAW @P: 0, B
3940 RDRAW @P: 0, -2*B
3950 RDRAW @P: 0, B
3960 RETURN
3970 REM ** SUB: DEPTH TICKMARKS AND VALUES, 4662 PLOT
3980 MOVE @1: 12, 10
3990 K5=1
4000 K6=6.5
4010 K7=0.4
4020 RMOVE @1: G1, 0
4030 GOSUB 4110
4040 FOR I=1 TO INT((Z0-Z9)/M1)
4050 RMOVE @1: M1*C1, 0
4060 H1=H1+M1
4070 IF H1=>Z0 THEN 4100
4080 GOSUB 4110
4090 NEXT I
4100 RETURN
4110 REM ** SUB: TICKMARK PLOTTING
4120 RDRAW @1: 0, K7
4130 RMOVE @1: 0, B3-2*K7
4140 RDRAW @1: 0, K7
4150 RMOVE @1: 0, -B3
4160 RMOVE @1: K5, -K6
4170 PRINT @1: USING "3D": H1
4180 RMOVE @1: -K5, K6
4190 RETURN
4200 REM ** SUB: TIME TICKMARKS AND VALUES, 4662 PLOT

```

```

4210 MOVE @1:12,10
4220 K5=1
4230 K6=4.7
4240 RMOVE @1:0,G1
4250 GOSUB 4330
4260 FOR I=1 TO INT((T0-T9)/M2)
4270 RMOVE @1:0,M2*C1
4280 H1=H1+M2
4290 IF H1=>T0 THEN 4320
4300 GOSUB 4330
4310 NEXT I
4320 RETURN
4330 REM *** SUB: TICKMARK PLOTTING
4340 RDRAW @1:K7,0
4350 RDRAW @1:-K7,0
4360 RMOVE @1:-K5,-K6
4370 PRINT @1: USING "3D":H1
4380 RMOVE @1:K5,K6
4390 RETURN
4400 REM *** SUB: VELOCITY TICKMARKS AND VALUES, 4662 PLOT
4410 MOVE @1:122,10
4420 K5=3
4430 K6=-2.7
4440 RMOVE @1:0,G1
4450 GOSUB 4540
4460 FOR I=1 TO INT((V0-V9)/M3)
4470 H1=H1+M3
4480 IF H1=>V0-0.01 THEN 4520
4490 RMOVE @1:0,M3*C1
4500 GOSUB 4540
4510 NEXT I
4520 PRINT @1,7:
4530 RETURN
4540 REM *** SUB: TICKMARK PLOTTING
4550 RDRAW @1:-K7,0
4560 RDRAW @1:K7,0
4570 RMOVE @1:K5,K6
4580 PRINT @1: USING "D.D":H1
4590 RMOVE @1:-K5,-K6
4600 RETURN
4610 REM ** SUB: PLOT ARRIVAL TIME VS. DEPTH, 4662 PLOT
4620 WINDOW 0,20-29,0,T0-T9
4630 VIEWPORT B4,B5,B1,B2
4640 C=B0*(Z0-Z9)/B6
4650 B=B0*(T0-T9)/B3
4660 T2=T5-T9
4670 Z2=Z5-Z9
4680 FOR J=1 TO N1
4690 MOVE @1:S1(J)-Z9,0
4700 GOSUB 3890
4710 DRAW @1:Z2(J,12),T2(J,12)
4720 GOSUB 3890
4730 FOR K=N2 TO 3 STEP -1
4740 RDRAW @1:-S2(J),T2(J,K-1)-T2(J,K)
4750 GOSUB 3890
4760 NEXT K
4770 NEXT J
4780 RETURN
4790 REM ** SUB: PLOT INTERVAL VELOCITIES, 4662 PLOT
4800 WINDOW 0,20-29,0,V0-V9
4810 V2=V1-V9
4820 MOVE @1:Z2(N1,2),V2(N1,2)
4830 FOR J=N1 TO 1 STEP -1
4840 DRAW @1:Z2(J,2),V2(J,2)
4850 RDRAW @1:S2(J),0
4860 FOR K=2 TO N2-2
4870 RDRAW @1:0,V2(J,K+1)-V2(J,K)
4880 RDRAW @1:S2(J),0
4890 NEXT K

```

```

4900 NEXT J
4910 RDRAW @1:0,V2(1,12)-V2(1,11)
4920 RDRAW @1:S1(1)-Z5(1,12),0
4930 VIEWPORT 0,150,0,100
4940 MOVE @1:150,100
4950 RETURN
4960 PRINT "___          HEADER INFORMATION FROM FILE ",F1
4970 PRINT "NUMBER OF CABLE POSITIONS (N1) =";N1
4980 PRINT "NUMBER OF SEIS/CABLE (N2) =";N2
4990 PRINT "IDENTIFICATION NUMBER ON TYPE OF SURVEY (N3) = ";N3
5000 PRINT "  N3=3 indicates an inhole survey was run"
5010 PRINT "AREA (A$) = ";A$
5020 PRINT "HOLE DESIGNATION (U$) = ";U$
5030 PRINT "COUNTY AND STATE (C$) = ";C$
5040 PRINT "DATE DATA TAKEN (D$) = ";D$
5050 PRINT "LATITUDE (L$) = ";L$;" AND LONGITUDE (O$) = ";O$
5060 PRINT "DETECTOR TYPE (R$)= ";R$;" AND SOURCE TYPE (S$)= ";S$
5070 PRINT "ELEV OF TEST HOLE (E1) = ";E1;" ELEV OF ";
5080 PRINT "CHECK SEIS (E2) = ";E2
5090 PRINT "CHECK SEIS OFFSET (O1) = ";O1
5100 IF F1=0 THEN 5140
5110 DIM R1(N1),S1(N1),T5(N1,N2),Z1(N1,N2)
5120 FIND @2:F1
5130 READ @2:N1,N2,N3,A$,C$,D$,L$,O$,R$,S$,U$,E1,E2,O1,R1,S1,T5,Z1
5140 PRINT "___CABLE NO.      CHECK SHOT TIME      SHOT DEPTH"
5150 IMAGE 2X,2D,12X,3D,2D,12X,3D,D
5160 PRINT "      J              T5(J,1)              S1(J)"
5170 FOR J=1 TO N1
5180 PRINT USING 5150:J,T5(J,1),S1(J)
5190 NEXT J
5200 FOR J=1 TO N1
5210 PRINT "___FOR CABLE POSITION ";J;" ":"
5220 PRINT "SEIS NO. "," " ARRIVAL TIME"; " SEIS DEPTH"
5230 PRINT "      K              T5(";J;"",K)              Z5(";J;"",K)"
5240 FOR K=1 TO N2
5250 IMAGE 2X,2D,8X,4D,2D,9X,3D,D
5260 PRINT USING 5250:K,T5(J,K),Z5(J,K)
5270 NEXT K
5280 NEXT J
5290 MOVE 0,0
5300 PRINT
5310 RETURN
5320 REM ** SUB: CHANGE ARRIVAL TIMES
5330 PRINT "___DO YOU WANT TO CHANGE ANY ARRIVAL TIMES? (Y OR N) ";
5340 INPUT G$
5350 IF G$="N" THEN 5540
5360 PRINT "___ENTER VALUES TO CHANGE ARRIVAL TIMES"
5370 PRINT "      For detector number (ordered pair), see tabulated values"
5380 PRINT "___NO. OF DETECTOR WHOSE ARRIVAL TIME IS TO BE CORRECTED = ";
5390 INPUT J
5400 PRI "K_
5410 INPUT K
5420 PRINT "      For detector ";J;"",K;" at a depth of ";Z5(J,K);" ",
5430 PRINT "old arrival time = ";T5(J,K);" new arrival time is to = ";
5440 INPUT T5(J,K)
5450 PRINT "___DO YOU WANT TO CHANGE OTHER ARRIVALS? (Y OR N) ";
5460 INPUT G$
5470 IF G$="N" THEN 5490
5480 GO TO 5380
5490 IF F1=0 THEN 5540
5500 PRINT "___DO YOU WANT TO STORE CHANGED VALUES? (Y OR N) ";
5510 INPUT G$
5520 IF G$="N" THEN 5540
5530 GOSUB 1170
5540 RETURN

```

THE CROSSHOLE SURVEY PROGRAM

A crosshole (hole-to-hole) survey is one in which seismic sources and detectors are positioned within different holes. Crosshole surveys have objectives dependent upon the separation between the source and receiver holes: when the two holes are close together, the purpose is to measure horizontal seismic velocities; when the holes are widely separated, the main objective is to search for the existence of seam waves (also called trapped or channel waves). This program assumes a field procedure in which a string of detectors is vertically deployed in one hole and a single source is located in another hole.

Figure 30 shows a field setup for conducting a crosshole survey using one source position and two cable positions. Note that the upper detector of the cable in its first position is at the same depth as the lower detector on the cable in its second position.

Straight-ray velocity is computed by dividing the distance between the source and a detector by the first-arrival time to that detector.

In the sample problem that follows, the section from a depth of 91 to 102 m is examined using one cable position. Separation between each of the 12 velocity-cable detectors is 1 m, and the hole-to-hole separation is 10 m. First-arrival times are entered to the nearest 0.01 msec in the sample.

Figure 31 is a copy of the screen display showing the first page of information required by the crosshole survey program. A copy of the screen display showing how depth and arrival time data are entered is given on figure 32.

Figure 33 is a copy of the screen display showing the third page of information required by the crosshole program in this sample problem. Here,




Figure 30. Field setup for conducting a hole-to-hole (crosshole) survey.

YOU HAVE SELECTED PROGRAM TO ENTER, COMPUTE,
DISPLAY, AND STORE HOLE-TO-HOLE SURVEY DATA

DO YOU WANT TO COMPUTE AND PLOT THOSE DATA STORED
ON A HOLE-TO-HOLE DATA STORAGE TAPE? (Y OR N) ☐ N

Area name (20 characters,max) = Watkins test area
Detector hole designation (7 characters,max) = 79-101A
Date on which survey was made = Feb. 14, 1980

Latitude of test hole = 39 48 45.6
Longitude of test hole = 103 23 12.3
North distance from detector hole to shotpoint hole = 6
East distance from detector hole to shotpoint hole = 8
County and State (23 characters,max) = Adams County, Colorado

Elevation of detector hole = 1650
Elevation of shotpoint hole = 1651

Type of in-hole detector used (16 characters,max) Piezoelectric
Type of seismic source (12 characters,max) Shotgun

Number of detectors on cable = 12
Number of inhole cable positions = 1
Depth of shotpoint = 98

Figure 31. Copy of screen display showing first page of information required by the crosshole survey program. In this example, only 1 inhole cable position was employed.

DETECTOR NUMBER	DETECTOR DEPTH	ARRIVAL TIME
1,1	91	5.71
1,2	92	5.19
1,3	93	4.71
1,4	94	4.82
1,5	95	4.65
1,6	96	5.05
1,7	97	5.45
1,8	98	5.58
1,9	99	5.48
1,10	100	5.39
1,11	101	5.36
1,12	102	5.21

Figure 32. Copy of screen display showing second page of information required by the crosshole survey program. Data to be entered are enclosed in a box.

DO YOU WANT TO STORE OBSERVED DATA ON A TAPE? <Y OR N> ☒ Y

INSERT HOLE-TO-HOLE DATA TAPE WITHIN 4924

WITHIN WHICH FILE ARE DATA TO BE STORED? ☐ 27

DO YOU WANT TO SEE TAPE OR ENTERED DATA? <Y OR N> ☒ Y

HEADER INFORMATION FROM FILE 27
NO. OF CABLE POSITIONS = 1; NO. OF SEIS/CABLE = 12
IDENTIFICATION NUMBER ON TYPE OF SURVEY (N3) = 4
N3=4 indicates a hole-to-hole survey was run
AREA = Watkins test area; HOLE DESIGNATION = 79-101A
CO. AND STATE = Adams County, Colorado; DATA DATE = Feb. 14, 1980
LATITUDE = 39 48 45.6 AND LONGITUDE = 103 23 12.3
DETECTOR TYPE = Piezoelectric AND SOURCE TYPE = Shotgun
ELEV OF TEST HOLE = 1650; ELEV OF SP HOLE = 1651
SP TO TEST HOLE DIST = 10; SP DEPTH = 98

SEIS NO.	FOR CABLE POSITION 1	ARRIVAL TIME	SEIS DEPTH
1	5.71	91.0	
2	5.19	92.0	
3	4.71	93.0	
4	4.82	94.0	
5	4.65	95.0	
6	5.05	96.0	
7	5.45	97.0	
8	5.58	98.0	
9	5.48	99.0	
10	5.39	100.0	
11	5.36	101.0	
12	5.21	102.0	

Figure 33. Copy of screen display showing third page of information required by the crosshole survey program. In this example, data entered from the keyboard were stored in file 27, and then a printout of the contents of that file was called by entering a "Y" to the third question.

as in previous examples, we have stored the entered data on tape (file 27) and then asked to see those data printed.

Figure 34 is a copy of the screen display produced by the crosshole survey program after plot limits, tickmark intervals, and choice of plotter has been made. Here we have chosen to plot on the screen.

Figure 35 is a copy of the screen display showing the plot of the results obtained with the cross hole survey program using information entered in figures 31 through 34. Note that the range in arrival time is only slightly more than 1 msec.

A copy of the screen display showing a tabulation of observed data and computed results is shown in figure 36.

Figure 37 is a copy of the screen display when information is called from file 27 and a plot on the 4662 plotter is to be made using the plot parameters as shown within the boxes.

Figures 38 is a reduced copy produced by the 4662 plotter using the information shown in figure 37 and the data stored in file 27.

ENTER PLOT-LIMIT VALUES

Min. depth of detector on plot =
Max. depth of detector on plot =

Min. arrival time on plot =
Max. arrival time on plot =

Min. velocity on plot =
Max. velocity on plot =

TICKMARK INTERVALS ON PLOT

Depth tickmark interval =
Arrival-time tickmark interval =
Velocity tickmark interval =

DO YOU WANT TO PLOT ON THE 4662? (Y OR N)

Figure 34. Copy of screen display produced by the crosshole survey program when it asked for plot limits, label intervals, and choice of plotter. In this case, plotting on the screen was selected.

HOLE-TO-HOLE SURVEY:79-101A AREA:Watkins test area DATE:Feb.14,1988
 LAT: 39 48 45.6; LONG: 103 23 12.3; Adams County, Colorado FILE NO: 27
 HOLE ELEV=1650.0 SP-HOLE ELEV=1651.0 SP DEPTH= 98.0 SP-HOLE O/S= 10.0
 DETECTOR:Piezoelectric SEISMIC SOURCE:Shotgun




Figure 35. Copy of screen display showing plot of results obtained with the crosshole survey program using information entered in figures 31 through 34. Straight-ray velocity equals the direct distance from the source to the detector divided by the arrival time at that detector.

HOLE-TO-HOLE SURVEY:79-101A AREA:Watkins test area DATE:Feb.14,1980
 LAT: 39 48 45.6; LONG: 103 23 12.3; Adams County, Colorado FILE NO: 27
 HOLE ELEV=1650.0 SP-HOLE ELEV=1651.0 SP DEPTH=98.0 SP-HOLE Q/S= 10.0
 DETECTOR:Piezoelectric SEISMIC SOURCE:Shotgun

DETECTORS NO. DEPTH	SP-DETECTOR DIST. TIME	STRAIGHT-RAY VELOCITY	REMARKS
1, 1 91.0	11.7 5.71	2.04	
1, 2 92.0	11.2 5.19	2.15	
1, 3 93.0	10.8 4.71	2.29	
1, 4 94.0	10.4 4.82	2.17	
1, 5 95.0	10.2 4.65	2.19	
1, 6 96.0	10.0 5.05	1.99	
1, 7 97.0	10.0 5.45	1.83	
1, 8 98.0	10.0 5.58	1.80	
1, 9 99.0	10.2 5.48	1.86	
1, 10 100.0	10.4 5.39	1.94	
1, 11 101.0	10.8 5.36	2.01	
1, 12 102.0	11.2 5.21	2.15	

DO YOU WANT TO ADJUST ANY ARRIVAL TIMES? (Y OR N) N

PROGRAM COMPLETED

Figure 36. Copy of screen display showing tabulation of observed data and computed results.

YOU HAVE SELECTED PROGRAM TO ENTER, COMPUTE,
DISPLAY, AND STORE HOLE-TO-HOLE SURVEY DATA

DO YOU WANT TO COMPUTE AND PLOT THOSE DATA STORED
ON A HOLE-TO-HOLE DATA STORAGE TAPE? (Y OR N) ☒ Y

INSERT INHOLE SURVEY DATA TAPE WITHIN 4924

WITHIN WHICH FILE ARE DATA STORED? ☐ 27

DO YOU WANT TO SEE TAPE OR ENTERED DATA? (Y OR N) ☒ H

ENTER PLOT-LIMIT VALUES

Min. depth of detector on plot = ☐ 90
Max. depth of detector on plot = ☐ 103

Min. arrival time on plot = ☐ 4.5
Max. arrival time on plot = ☐ 6

Min. velocity on plot = ☐ 1.5
Max. velocity on plot = ☐ 2.5

TICKMARK INTERVALS ON PLOT

Depth tickmark interval = ☐ 1
Arrival-time tickmark interval = ☐ .2
Velocity tickmark interval = ☐ .2

DO YOU WANT TO PLOT ON THE 4662? (Y OR N) ☒ Y

Figure 37. Copy of screen display showing information required by the crosshole survey program when a display on the 4662 plotter is to be made. In this case, data was called from file 27.




Figure 38. Reduced copy of report-quality plot produced by the 4662 plotter after information of figure 37 was entered. Original dimensions of the outer border were 25.4 by 33.0 cm. Information needed to write a figure caption can be obtained either from the 4-line label printed across the top of screen display (fig. 35) or data tabulation (fig. 36).

Listing of program for crosshole-survey data

```

100 PRINT "L_ YOU HAVE SELECTED PROGRAM TO ENTER, COMPUTE, "
110 PRINT " DISPLAY, AND STORE HOLE-TO-HOLE SURVEY DATA"
120 INIT
130 DIM B$(24), G$(1), H$(19), N$(15), U$(7)
140 DATA 0, 25, 0, 5, 0, 5, 32, "DEPTH IN METERS", "HOLE-TO-HOLE SURVEY"
150 READ B0, C, B, P, N$, H$
160 B$="ARRIVAL TIME IN MILLISEC"
170 PRINT "___DO YOU WANT TO COMPUTE AND PLOT THOSE DATA STORED"
180 PRINT "ON A HOLE-TO-HOLE DATA STORAGE TAPE? (Y OR N) ";
190 INPUT G$
200 IF G$="N" THEN 340
210 PRINT "G_G_G___INSERT INHOLE SURVEY DATA TAPE WITHIN 4924"
220 PRINT "___WITHIN WHICH FILE ARE DATA STORED? ";
230 INPUT F1
240 FIND @2:F1
250 READ @2:N1, N2, N3
260 IF N3=4 THEN 290
270 PRINT "G_G_G___ERROR--YOU HAVE NOT SELECTED A HOLE-TO-HOLE DATA SET"
280 GO TO 170
290 DIM S1(1), T5(N1, N2), Z5(N1, N2)
300 FIND @2:F1
310 READ @2:N1, N2, N3, A$, C$, D$, L$, O$, R$, S$, U$, E1, E2, O1, S1, S1, S1, T5, Z5
320 S=S1(1)
330 GO TO 360
340 F1=0
350 GOSUB 550
360 PRINT "___DO YOU WANT TO SEE TAPE OR ENTERED DATA? (Y OR N) ";
370 INPUT G$
380 IF G$="N" THEN 400
390 GOSUB 4880
400 GOSUB 1130
410 GOSUB 1350
420 GOSUB 1640
430 GOSUB 2240
440 GOSUB 2560
450 GOSUB 2850
460 GOSUB 3110
470 DELETE T2
480 GOSUB 3310
490 GOSUB 3420
500 IF P=1 THEN 530
510 GOSUB 3630
520 GOSUB 5080
530 PRINT "G_G_G___PROGRAM COMPLETED"
540 END
550 REM *** SUB:ENTER SURVEY DESCRIPTION DATA
560 N3=4
570 PRINT "___Area name (20 characters,max) = ";
580 INPUT A$
590 PRINT "Detector hole designation (7 characters,max) = ";
600 INPUT U$
610 PRINT "Date on which survey was made = ";
620 INPUT D$
630 PRINT "___Latitude of test hole = ";
640 INPUT L$
650 PRINT "Lonsitude of test hole = ";
660 INPUT O$
670 PRINT "North distance from detector hole to shotPoint hole = ";
680 INPUT D4
690 PRINT "East distance from detector hole to shotPoint hole = ";
700 INPUT D5
710 O1=SQR(D4*D4+D5*D5)
720 PRINT "County and State (23 characters,max) = ";
730 INPUT C$
740 PRINT "___Elevation of detector hole = ";
750 INPUT E1

```

```

760 PRINT "Elevation of shotPoint hole = ";
770 INPUT E2
780 PRINT "___Type of in-hole detector used (16 characters,max) ";
790 INPUT R$
800 PRINT "Type of seismic source (12 characters,max) ";
810 INPUT S$
820 PRINT "___Number of detectors on cable = ";
830 INPUT N2
840 PRINT "Number of inhole cable positions = ";
850 INPUT N1
860 DIM S1(1), T5(N1,N2), Z5(N1,N2)
870 PRINT "Depth of shotPoint = ";
880 INPUT S1(1)
890 S=S1(1)
900 FOR J=1 TO N1
910 PRINT "L_DETECTOR    DETECTOR    ARRIVAL"
920 PRINT "  NUMBER      DEPTH      TIME"
930 FOR K=1 TO N2
940 PRINT "___ "; J; ", "; K; "          ";
950 INPUT Z5(J,K)
960 PRINT "K_          ";
970 INPUT T5(J,K)
980 NEXT K
990 MOVE 0,0
1000 PRINT "_____"
1010 NEXT J
1020 PRINT "___DO YOU WANT TO STORE OBSERVED DATA ON A TAPE?";
1030 PRINT " (Y OR N ) ";
1040 INPUT G$
1050 IF G$="N" THEN 1120
1060 PRINT "G_G_G___INSERT HOLE-TO-HOLE DATA TAPE WITHIN 4924"
1070 PRINT "___WITHIN WHICH FILE ARE DATA TO BE STORED? ";
1080 INPUT F1
1090 FIND @2:F1
1100 WRITE @2:N1,N2,N3,A$,C$,D$,L$,O$,R$,S$,U$,E1,E2,O1,S1,S1,S1,T5,Z5
1110 PRINT @2,2:
1120 RETURN
1130 REM *** SUB: ENTER PLOT LIMITS
1140 PRINT "___ENTER PLOT-LIMIT VALUES"
1150 PRINT "___  Min. depth of detector on plot = ";
1160 INPUT Z9
1170 PRINT "      Max. depth of detector on plot = ";
1180 INPUT Z0
1190 PRINT "___  Min. arrival time on plot = ";
1200 INPUT T9
1210 PRINT "      Max. arrival time on plot = ";
1220 INPUT T0
1230 PRINT "___  Min. velocity on plot = ";
1240 INPUT V9
1250 PRINT "      Max. velocity on plot = ";
1260 INPUT V0
1270 PRINT "___TICKMARK INTERVALS ON PLOT"
1280 PRINT "___  Depth tickmark interval = ";
1290 INPUT M1
1300 PRINT "      Arrival-time tickmark interval = ";
1310 INPUT M2
1320 PRINT "      Velocity tickmark interval = ";
1330 INPUT M3
1340 RETURN
1350 REM *** SUB: PRINT SURVEY-INFORMATION DATA
1360 PRINT "___DO YOU WANT TO PLOT ON THE 4662? (Y OR N) ";
1370 INPUT G$
1380 IF G$="Y" THEN 1600
1390 MOVE 0,0
1400 PRINT

```



```

1410 DIM A$(20), C$(23), D$(12), L$(10), O$(11), R$(16), S$(12)
1420 DIM V$(9), W$(8), X$(10), Y$(9), Z$(15)
1430 RESTORE 1440
1440 DATA 1, 792, 2, 816, 9, 88, 9, 69, 70, 129
1450 READ K1, K2, B1, B2, B4, B5, B7, B8
1460 IMAGE "L-HOLE-TO-HOLE SURVEY:", 7A, " AREA:", 20A, " DATE:", 12A
1470 PRINT USING 1460: U$, A$, D$
1480 IMAGE "LAT: ", 10A, "; LONG: ", 11A, "; ", 23A, " FILE NO: ", 3D
1490 PRINT USING 1480: L$, O$, C$, F1
1500 V$="DETECTOR:"
1510 W$="SP-HOLE "
1520 X$="HOLE ELEV="
1530 Y$="SP DEPTH="
1540 Z$="SEISMIC SOURCE:"
1550 IMAGE 10A, 4D, D, 2X, 8A, "ELEV=", 4D, D, 2X, 9A, 3D, D, 2X, 8A, "O/S=", 3D, D
1560 PRINT USING 1550: X$, E1, W$, E2, Y$, S, W$, O1
1570 IMAGE 9A, 16A, 20X, 15A, 12A
1580 PRINT USING 1570: V$, R$, Z$, S$
1590 RETURN
1600 RESTORE 1610
1610 DATA 1, 2, 3, 0, 5, 10, 95, 13, 122
1620 READ P, K3, K4, B0, B1, B2, B4, B5
1630 RETURN
1640 REM ** SUB: PLOT BORDERS AND LABEL AXES
1650 WINDOW 0, 130, 0, 100
1660 VIEWPORT 0, 130, 0, 100
1670 IF P=1 THEN 1700
1680 B9=B8-B7
1690 GO TO 1760
1700 MOVE @1:0,0
1710 RDRAW @1:0,100
1720 RDRAW @1:130,0
1730 RDRAW @1:0,-100
1740 RDRAW @1:-130,0
1750 B9=B2-B1
1760 B3=B2-B1
1770 B6=B5-B4
1780 MOVE @P:B4,B2
1790 RDRAW @P:B6,0
1800 RDRAW @P:0,-B3
1810 RDRAW @P:-B6,0
1820 RDRAW @P:0,B3
1830 IF P=1 THEN 1990
1840 MOVE B7,B2
1850 RDRAW B9,0
1860 RDRAW 0,-B3
1870 RDRAW -B9,0
1880 RDRAW 0,B3
1890 MOVE 0,(B1+B2+(LEN(N$)-1.5)*K2)/2
1900 FOR I=1 TO LEN(N$)
1910 F$=SEG(N$,I,1)
1920 PRINT F$
1930 NEXT I
1940 MOVE (B4+B5)/2-12*K1,B1-6
1950 PRINT B$, "K_"
1960 MOVE (B7+B8-31*K1)/2,B1-6
1970 PRINT "STRAIGHT-RAY VELOCITY IN M/MSEC", "K_"
1980 RETURN
1990 MOVE @1:9,22
2000 PRINT @1,25:90
2010 PRINT @1,17:K3,K4
2020 PRINT @1:B$
2030 MOVE @1:8,84
2040 FOR J=1 TO 5
2050 RMOVE @1:0,-2

```

```

2060 GOSUB 3840
2070 NEXT J
2080 MOVE @1:128,11
2090 PRINT @1:"STRAIGHT-RAY VELOCITY IN M/MILLISEC"
2100 MOVE @1:127,96
2110 FOR J=1 TO 5
2120 RMOVE @1:0,-2
2130 GOSUB 3760
2140 NEXT J
2150 MOVE @1:82,3
2160 PRINT @1,25:180
2170 PRINT @1:N$
2180 PRINT @1,17:2,7,4
2190 PRINT @1,25:90
2200 MOVE @1:4,3
2210 PRINT @1:"HOLE-TO-HOLE SURVEY AT HOLE ";U$
2220 PRINT @1,17:K3,K4
2230 RETURN
2240 REM ** SUB: PLOT AND LABEL DEPTH TICKMARKS
2250 IF P=32 THEN 2280
2260 C1=B6/(Z0-Z9)
2270 GO TO 2290
2280 C1=B3/(Z0-Z9)
2290 H1=M1*(INT(Z9/M1)+1)
2300 G1=(H1-Z9)*C1
2310 IF P=32 THEN 2340
2320 GOSUB 3920
2330 RETURN
2340 MOVE B4,B2
2350 RMOVE 0,-G1
2360 GOSUB 2440
2370 FOR I=1 TO INT((Z0-Z9)/M1)
2380 RMOVE 0,-M1*C1
2390 H1=H1+M1
2400 IF H1>Z0 THEN 2430
2410 GOSUB 2440
2420 NEXT I
2430 RETURN
2440 REM ** SUB: TICKMARK PLOTTING
2450 RDRAW 0,2,0
2460 RMOVE B6-0,4,0
2470 RDRAW 0,2,0
2480 RMOVE 1,0
2490 RDRAW 0,2,0
2500 RMOVE B9-0,4,0
2510 RDRAW 0,2,0
2520 RMOVE -B6-B9-6,4,-K2/2
2530 PRINT USING "3D":H1
2540 RMOVE 5,4,K2/2
2550 RETURN
2560 REM ** SUB: PLOT AND LABEL ARRIVAL-TIME TICKMARKS
2570 IF P=32 THEN 2600
2580 C1=B3/(T0-T9)
2590 GO TO 2610
2600 C1=B6/(T0-T9)
2610 H1=M2*(INT(T9/M2)+1)
2620 G1=(H1-T9)*C1
2630 IF P=32 THEN 2660
2640 GOSUB 4150
2650 RETURN
2660 MOVE B4,B1
2670 RMOVE G1,0
2680 GOSUB 2700
2690 GO TO 2780
2700 REM ** SUB: TICKMARK PLOTTING

```

```

2710 RDRAW 0,0.2
2720 RMOVE 0,B3-0.4
2730 RDRAW 0,0.2
2740 RMOVE -3.3*K1,-B3-K2
2750 PRINT @P: USING "3D.D":H1
2760 RMOVE 3.3*K1,K2
2770 RETURN
2780 FOR I=1 TO INT((T0-T9)/M2)
2790 H1=H1+M2
2800 IF H1>T0-1.0E-3 THEN 2840
2810 RMOVE M2*C1,0
2820 GOSUB 2700
2830 NEXT I
2840 RETURN
2850 REM ** SUB: PLOT AND LABEL VELOCITY TICKMARKS
2860 C1=B9/(V0-V9)
2870 H1=M3*(INT(V9/M3)+1)
2880 G1=(H1-V9)*C1
2890 IF P=32 THEN 2920
2900 GOSUB 4350
2910 RETURN
2920 MOVE B7,B1
2930 RMOVE G1,0
2940 GOSUB 2960
2950 GO TO 3040
2960 REM ** SUB: TICKMARK PLOTTING
2970 RDRAW 0,0.2
2980 RMOVE 0,B3-0.4
2990 RDRAW 0,0.2
3000 RMOVE -1.3*K1,-B3-K2
3010 PRINT USING "D.D":H1
3020 RMOVE 1.3*K1,K2
3030 RETURN
3040 FOR I=1 TO INT((V0-V9)/M3)
3050 H1=H1+M3
3060 IF H1>V0-1.0E-3 THEN 3100
3070 RMOVE M3*C1,0
3080 GOSUB 2960
3090 NEXT I
3100 RETURN
3110 REM ** SUB: PLOT HOLE-TO-HOLE ARRIVAL TIME VS. DEPTH
3120 DIM T2(N1,N2),Z2(N1,N2)
3130 IF P=32 THEN 3160
3140 GOSUB 4560
3150 RETURN
3160 WINDOW 0,T0-T9,0,Z0-Z9
3170 VIEWPORT B4,B5,B1,B2
3180 C=B0*(T0-T9)/B6
3190 B=B0*(Z0-Z9)/B3
3200 T2=T5-T9
3210 Z2=Z0-Z5
3220 FOR J=1 TO N1
3230 MOVE T2(J,1),Z2(J,1)
3240 GOSUB 3840
3250 FOR K=2 TO N2
3260 RDRAW T2(J,K)-T2(J,K-1),Z2(J,K)-Z2(J,K-1)
3270 GOSUB 3840
3280 NEXT K
3290 NEXT J
3300 RETURN
3310 REM ** SUB: COMPUTE STRAIGHT-RAY VELOCITIES
3320 DIM D3(N1,N2),V1(N1,N2)
3330 S2=S+E1-E2
3340 FOR J=1 TO N1
3350 FOR K=1 TO N2

```

```

3360 D3(J,K)=Z5(J,K)-S2
3370 D3(J,K)=SQR(D3(J,K)*D3(J,K)+01*01)
3380 V1(J,K)=D3(J,K)/T5(J,K)
3390 NEXT K
3400 NEXT J
3410 RETURN
3420 REM ** SUB: PLOT STRAIGHT-RAY VELOCITIES
3430 DIM V2(N1,N2)
3440 IF P=32 THEN 3470
3450 GOSUB 4720
3460 RETURN
3470 WINDOW 0,V0-V9,0,Z0-Z9
3480 VIEWPORT B7,B8,B1,B2
3490 C=B0*(V0-V9)/B9
3500 V2=V1-V9
3510 Z2=Z0-Z5
3520 FOR J=1 TO N1
3530 MOVE V2(J,1),Z2(J,1)
3540 GOSUB 3760
3550 FOR K=2 TO N2
3560 RDRAW V2(J,K)-V2(J,K-1),Z2(J,K)-Z2(J,K-1)
3570 GOSUB 3760
3580 NEXT K
3590 NEXT J
3600 MOVE 0,0
3610 PRINT "-----"
3620 RETURN
3630 REM ** SUB: TABULATE RESULTS
3640 GOSUB 1460
3650 PRINT "---- DETECTORS      SP-DETECTOR      ";
3660 PRINT "STRAIGHT-RAY      REMARKS"
3670 PRINT " NO. DEPTH      DIST. TIME      VELOCITY__"
3680 IMAGE 2D, " ", 2D, 2X, 3D, D, 4X, 3D, D, X, 3D, 2D, 8X, 2D, 2D
3690 FOR J=1 TO N1
3700 FOR K=1 TO N2
3710 PRINT USING 3680: J, K, Z5(J,K), D3(J,K), T5(J,K), V1(J,K)
3720 NEXT K
3730 PRINT
3740 NEXT J
3750 RETURN
3760 REM ** SUB: CROSS SYMBOL
3770 RDRAW @P:-C,0
3780 RDRAW @P:2*C,0
3790 RDRAW @P:-C,0
3800 RDRAW @P:0,B
3810 RDRAW @P:0,-2*B
3820 RDRAW @P:0,B
3830 RETURN
3840 REM ** SUB: X SYMBOL
3850 RDRAW @P:-C,B
3860 RDRAW @P:2*C,-2*B
3870 RDRAW @P:-C,B
3880 RDRAW @P:-C,-B
3890 RDRAW @P:2*C,2*B
3900 RDRAW @P:-C,-B
3910 RETURN
3920 REM ** SUB: DEPTH TICKMARKS AND VALUES, 4662 PLOT
3930 MOVE @1:13,10
3940 K5=1
3950 K6=6.5
3960 K7=0.4
3970 RMOVE @1:G1,0
3980 GOSUB 4060
3990 FOR I=1 TO INT((Z0-Z9)/M1)
4000 RMOVE @1:M1*C1,0

```

```

4010 H1=H1+M1
4020 IF H1=>Z0 THEN 4050
4030 GOSUB 4060
4040 NEXT I
4050 RETURN
4060 REM ** SUB: TICKMARK PLOTTING
4070 RDRAW @1:0,K7
4080 RMOVE @1:0,B3-2*K7
4090 RDRAW @1:0,K7
4100 RMOVE @1:0,-B3
4110 RMOVE @1:K5,-K6
4120 PRINT @1: USING "3D":H1
4130 RMOVE @1:-K5,K6
4140 RETURN
4150 REM ** SUB: TIME TICKMARKS AND VALUES, 4662 PLOT
4160 MOVE @1:13,10
4170 K5=1
4180 K6=6.7
4190 RMOVE @1:0,G1
4200 GOSUB 4280
4210 FOR I=1 TO INT((T0-T9)/M2)
4220 RMOVE @1:0,M2*C1
4230 H1=H1+M2
4240 IF H1=>T0 THEN 4270
4250 GOSUB 4280
4260 NEXT I
4270 RETURN
4280 REM ** SUB: TICKMARK PLOTTING
4290 RDRAW @1:K7,0
4300 RDRAW @1:-K7,0
4310 RMOVE @1:-K5,-K6
4320 PRINT @1: USING "3D.D":H1
4330 RMOVE @1:K5,K6
4340 RETURN
4350 REM ** SUB: VELOCITY TICKMARKS AND VALUES, 4662 PLOT
4360 MOVE @1:122,10
4370 K5=3
4380 K6=2.7
4390 RMOVE @1:0,G1
4400 GOSUB 4490
4410 FOR I=1 TO INT((V0-V9)/M3)
4420 H1=H1+M3
4430 IF H1=>V0 THEN 4470
4440 RMOVE @1:0,M3*C1
4450 GOSUB 4490
4460 NEXT I
4470 PRINT @1,7:
4480 RETURN
4490 REM ** SUB: TICKMARK PLOTTING
4500 RDRAW @1:-K7,0
4510 RDRAW @1:K7,0
4520 RMOVE @1:K5,-K6
4530 PRINT @1: USING "D.D":H1
4540 RMOVE @1:-K5,K6
4550 RETURN
4560 REM ** SUB: PLOT ARRIVAL TIME VS. DEPTH, 4662 PLOT
4570 WINDOW 0,20-29,0,T0-T9
4580 VIEWPORT B4,B5,B1,B2
4590 C=0.8*B0*(Z0-Z9)/B6
4600 B=0.8*B0*(T0-T9)/B3
4610 T2=T5-T9
4620 Z2=Z5-Z9
4630 FOR J=1 TO N1
4640 MOVE @1:Z2(J,1),T2(J,1)
4650 GOSUB 3840

```

```

4660 FOR K=2 TO N2
4670 RDRAW @1:Z2(J,K)-Z2(J,K-1), T2(J,K)-T2(J,K-1)
4680 GOSUB 3840
4690 NEXT K
4700 NEXT J
4710 RETURN
4720 REM ** SUB: PLOT STRAIGHT-RAY VELOCITY, 4662 PLOT
4730 WINDOW 0, Z0-Z9, 0, V0-V9
4740 V2=V1-V9
4750 B=B0*(V0-V9)/B3
4760 C=B0*(Z0-Z9)/B6
4770 FOR J=1 TO N1
4780 MOVE @1:Z2(J,1), V2(J,1)
4790 GOSUB 3760
4800 FOR K=2 TO N2
4810 RDRAW @1:Z2(J,K)-Z2(J,K-1), V2(J,K)-V2(J,K-1)
4820 GOSUB 3760
4830 NEXT K
4840 NEXT J
4850 VIEWPORT 0,150,0,100
4860 MOVE @1:150,100
4870 RETURN
4880 REM ** SUB: CONTENTS OF DATA STORAGE TAPE
4890 PRINT "--          HEADER INFORMATION FROM FILE ";F1
4900 PRINT "NO. OF CABLE POSITIONS = ";N1; " NO. OF SEIS/CABLE = ";N2
4910 PRINT "IDENTIFICATION NUMBER ON TYPE OF SURVEY (N3) = ";N3
4920 PRINT "    N3=4 indicates a hole-to-hole survey was run"
4930 PRINT "AREA = ";A$; " HOLE DESIGNATION = ";U$
4940 PRINT "CO. AND STATE = ";C$; " DATA DATE = ";D$
4950 PRINT "LATITUDE = ";L$; " AND LONGITUDE = ";O$
4960 PRINT "DETECTOR TYPE = ";R$; " AND SOURCE TYPE = ";S$
4970 PRINT "ELEV OF TEST HOLE = ";E1; " ELEV OF SP HOLE = ";E2
4980 PRINT "SP TO TEST HOLE DIST = ";O1; " SP DEPTH = ";S1(1)
4990 FOR J=1 TO N1
5000 PRINT "--          FOR CABLE POSITION ";J
5010 PRINT "SEIS NO. "; " ARRIVAL TIME"; " SEIS DEPTH"
5020 FOR K=1 TO N2
5030 IMAGE 2X, 2D, 8X, 4D, 2D, 9X, 3D, D
5040 PRINT USING 5030:K, T5(J,K), Z5(J,K)
5050 NEXT K
5060 NEXT J
5070 RETURN
5080 REM ** SUB: CHANGE ARRIVAL TIMES
5090 PRINT "--DO YOU WANT TO ADJUST ANY ARRIVAL TIMES? (Y OR N) ";
5100 INPUT G$
5110 IF G$="N" THEN 5280
5120 PRINT "--NO. OF DETECTOR WHOSE ARRIVAL TIME IS TO BE ADJUSTED = ";
5130 INPUT J
5140 PRI "K_";
5150 INPUT K
5160 PRINT "    For detector ";J; ", ";K; " at a depth of ";Z5(J,K); ", "
5170 PRINT "old arrival time = ";T5(J,K); "; new arrival time is to = ";
5180 INPUT T5(J,K)
5190 PRINT "--DO YOU WANT TO ADJUST OTHER ARRIVALS? (Y OR N) ";
5200 INPUT G$
5210 IF G$="N" THEN 5270
5220 GO TO 5120
5230 IF F1=0 THEN 5280
5240 PRINT "--DO YOU WANT TO STORE CHANGED VALUES? (Y OR N) ";
5250 INPUT G$
5260 IF G$="N" THEN 5280
5270 GOSUB 1090
5280 RETURN

```

THE DATA REVIEW AND EDIT PROGRAM

The program to view or change contents of a data file can be applied to a downhole, inhole, or crosshole survey data file. Only one question has to be answered to begin running this program, namely:

WHICH FILE DO YOU WISH TO EXAMINE?

Upon receiving the file number, the program finds and prints the contents of that file. In these tabulations, the names of the string and numeric variables are also printed. For example, A\$ is the string variable for the area name and T5(1, 2) is the numeric variable for the arrival time of the second detector on cable 1. These names are needed in the editing process.

Figures 39 and 40 are copies of the screen displays showing the labeled contents of file 25, the file containing the downhole-survey sample problem.

Figures 41 through 43 are copies of the screen display showing the labeled contents of 26, the file containing the inhole-survey sample problem.

Figure 44 is a copy of the screen display showing the contents of file 27, the file containing the crosshole-survey sample problem.

As an exercise, let us assume that we want to change some of the information in file 27. To do so, we would enter a "Y" in response to the question across the top of figure 45. The edit program would then print the instruction for making changes. Those quantities that were changed in this example are enclosed within the box. After the new values were entered, the command RUN 1120 was given. Thereupon, the program wrote the revised information on the same file from which the original values were read.

Figure 46 is a copy of the screen display showing the contents of file 27 after the change has been made. On this figure a dashed-line box has been drawn around those values that had been changed.

HEADER INFORMATION FROM FILE 25
 NUMBER OF CABLE POSITIONS (N1) = 3
 NUMBER OF SEIS/CABLE (N2) = 11
 IDENTIFICATION NUMBER ON TYPE OF SURVEY (N3) = 1
 A downhole survey using a velocity cable was run
 AREA (A\$) = Watkins test area
 HOLE DESIGNATION (U\$) = 79-101A
 COUNTY AND STATE (C\$) = Adams County, Colorado
 DATE DATA TAKEN (D\$) = Feb. 11, 1980
 LATITUDE (L\$) = 39 48 45.6 AND LONGITUDE (O\$) = 103 23 12.3
 DETECTOR TYPE (R\$) = Piezoelectric AND SOURCE TYPE (S\$) = Kinopak, 1/3#
 ELEV OF TEST HOLE (E1) = 1650; ELEV OF SP HOLE (E2) = 1650.2
 SP TO TEST HOLE DIST (O1) = 5

CABLE NO.	REFERENCE TIME	SHOT DEPTH	UPHOLE TIME
J	R1(J)	S1(J)	T1(J)
1	80.40	3.0	10.4
2	81.10	2.9	9.8
3	81.70	2.8	9.6

FOR CABLE POSITION 1:

SEIS NO.	ARRIVAL TIME	SEIS DEPTH
J, K	T5(1, K)	Z1(1, K)
1, 1	67.800	90.0
1, 2	70.000	94.0
1, 3	72.100	98.0
1, 4	74.000	102.0
1, 5	75.700	106.0
1, 6	77.100	110.0
1, 7	78.500	114.0
1, 8	79.800	118.0
1, 9	81.000	122.0
1, 10	82.100	126.0
1, 11	83.200	130.0

FOR CABLE POSITION 2:

Figure 39. Copy of screen display produced by the change program when used to recall and print contents of file 25. These data are those used in the downhole-survey sample problems.

SEIS NO.	ARRIVAL TIME	SEIS DEPTH
J, K	T5(2,K)	Z1(2,K)
2, 1	51.700	50.0
2, 2	53.600	54.0
2, 3	55.600	58.0
2, 4	56.900	62.0
2, 5	58.400	66.0
2, 6	60.000	70.0
2, 7	61.700	74.0
2, 8	63.400	78.0
2, 9	64.900	82.0
2, 10	66.400	86.0
2, 11	68.000	90.0

FOR CABLE POSITION 3:

SEIS NO.	ARRIVAL TIME	SEIS DEPTH
J, K	T5(3,K)	Z1(3,K)
3, 1	23.000	10.0
3, 2	28.700	14.0
3, 3	34.700	18.0
3, 4	37.400	22.0
3, 5	39.500	26.0
3, 6	41.600	30.0
3, 7	44.000	34.0
3, 8	46.200	38.0
3, 9	48.200	42.0
3, 10	50.200	46.0
3, 11	52.100	50.0

DO YOU WANT TO CHANGE ANY OF THESE DATA? (Y OR N)

PROGRAM COMPLETED

Figure 40. Copy of screen display showing second page of print of contents of file 25.

HEADER INFORMATION FROM FILE 26

NUMBER OF CABLE POSITIONS (N1) = 4
 NUMBER OF SEIS/CABLE (N2) = 12
 IDENTIFICATION NUMBER ON TYPE OF SURVEY (N3) = 3
 An inhole survey (source and detectors in same hole) was run
 AREA (A\$) = Watkins test area
 HOLE DESIGNATION (U\$) = 79-101A
 COUNTY AND STATE (C\$) = Adams County, Colorado
 DATE DATA TAKEN (D\$) = Feb. 13, 1980
 LATITUDE (L\$) = 39 48 45.6 AND LONGITUDE (O\$) = 103 23 12.3
 DETECTOR TYPE (R\$) = Piezoelectric AND SOURCE TYPE (S\$) = Shotgun
 ELEV OF TEST HOLE (E1) = 1650; ELEV OF CHECK SEIS (E2) = 1650.2
 CHECK SEIS OFFSET = 2

CABLE NO.	CHECK	SHOT TIME	SHOT DEPTH
J	T5(J,1)	S1(J)	
1	87.50	113.0	
2	83.80	103.0	
3	79.10	93.0	
4	74.70	83.0	

FOR CABLE POSITION 1:

SEIS NO.	ARRIVAL TIME	SEIS DEPTH
J, K	T5(1,K)	Z1(1,K)
1, 1	87.500	-0.2
1, 2	6.000	98.0
1, 3	5.440	99.0
1, 4	4.940	100.0
1, 5	4.490	101.0
1, 6	4.070	102.0
1, 7	3.690	103.0
1, 8	3.300	104.0
1, 9	2.920	105.0
1, 10	2.530	106.0
1, 11	2.150	107.0

Figure 41. Copy of screen display produced by the change program when used to recall and print contents of file 26. These data are those used in the inhole-survey sample problem.

1,12	FOR CABLE POSITION 2:	1.760	108.0
SEIS NO.	ARRIVAL TIME		SEIS DEPTH
J, K	T5(2,K)		Z1(2,K)
2, 1	83.800		-0.2
2, 2	7.180		88.0
2, 3	6.800		89.0
2, 4	6.410		90.0
2, 5	5.860		91.0
2, 6	5.300		92.0
2, 7	4.750		93.0
2, 8	4.360		94.0
2, 9	3.980		95.0
2, 10	3.420		96.0
2, 11	2.870		97.0
2, 12	2.310		98.0
	FOR CABLE POSITION 3:		
SEIS NO.	ARRIVAL TIME		SEIS DEPTH
J, K	T5(3,K)		Z1(3,K)
3, 1	79.100		-0.2
3, 2	6.280		78.0
3, 3	5.900		79.0
3, 4	5.510		80.0
3, 5	5.130		81.0
3, 6	4.740		82.0
3, 7	4.360		83.0
3, 8	3.970		84.0
3, 9	3.590		85.0
3, 10	3.210		86.0
3, 11	2.820		87.0
3, 12	2.440		88.0
	FOR CABLE POSITION 4:		
SEIS NO.	ARRIVAL TIME		SEIS DEPTH
J, K	T5(4,K)		Z1(4,K)
4, 1	74.700		-0.2

Figure 42. Copy of screen display showing second page of print of contents of file 26.

4, 3	6.050	68.0
4, 3	5.650	69.0
4, 4	5.250	70.0
4, 5	4.850	71.0
4, 6	4.450	72.0
4, 7	4.050	73.0
4, 8	3.650	74.0
4, 9	3.250	75.0
4, 10	2.690	76.0
4, 11	2.310	77.0
4, 12	1.920	78.0

DO YOU WANT TO CHANGE ANY OF THESE DATA? (Y OR N)

PROGRAM COMPLETED

Figure 43. Copy of screen display showing third page of contents of file 26.

HEADER INFORMATION FROM FILE 27

NUMBER OF CABLE POSITIONS (N1) = 1
 NUMBER OF SEIS/CABLE (N2) = 12
 IDENTIFICATION NUMBER ON TYPE OF SURVEY (N3) = 4
 A crosshole or hole-to-hole survey was run
 AREA (A\$) = Watkins test area
 HOLE DESIGNATION (U\$) = 79-101A
 COUNTY AND STATE (C\$) = Adams County, Colorado
 DATE DATA TAKEN (D\$) = Feb. 14, 1980
 LATITUDE (L\$) = 39 48 45.6 AND LONGITUDE (O\$) = 103 23 12.3
 DETECTOR TYPE (R\$) = Piezoelectric AND SOURCE TYPE (S\$) = Shotgun
 ELEV OF TEST HOLE (E1) = 1650; ELEV OF SP HOLE (E2) = 1651
 SP TO TEST HOLE DIST (O1) = 10

CABLE NO.	DEPTH OF SHOT	DEPTH UPPER DETECTOR	DEPTH LOWER DETECTOR
J	S1(1)	Z1(J,1)	Z1(J,12)
1	98.0	91.0	102.0

FOR CABLE POSITION 1:

SEIS NO.	ARRIVAL TIME	SEIS DEPTH
J, K	T5(1,K)	Z1(1,K)
1, 1	5.710	91.0
1, 2	5.190	92.0
1, 3	4.710	93.0
1, 4	4.820	94.0
1, 5	4.650	95.0
1, 6	5.050	96.0
1, 7	5.450	97.0
1, 8	5.580	98.0
1, 9	5.480	99.0
1, 10	5.390	100.0
1, 11	5.360	101.0
1, 12	5.210	102.0

Figure 44. Copy of screen display produced by the change program when used to recall and print contents of file 27. These data are those used in the crosshole (hole-to-hole) sample problem.

DO YOU WANT TO CHANGE ANY OF THESE DATA? (Y OR N) ☒ Y

PROCEDURE TO MAKE CHANGES IS AS FOLLOWS:

1. Identify name of variable to be changed; for example, A\$, T1(2), Z1(2,4)
2. Enter variable name and its new value; for example, A\$="WATKINS", T1(2)=12.5, Z1(2,4)=30
3. After all changes have been made, enter RUN 1120

ARE YOU READY TO PROCEED? (Y OR N) ☒ Y

```
A$="WATKINS"  
E2=1651.2  
S1(1)=98.2  
T5(1,12)=5.22  
Z1(1,12)=102.1
```

RUN 1120

PROGRAM COMPLETED

Figure 45. Copy of screen display showing second page of print of contents of file 27. (See Fig. 44.)

In this example, the choice was made to change data--a "Y" was entered in response to the question at the top of the page. Upon receiving this reply, the program prints the procedure to be followed, and then asks if you are ready to proceed. Values of the variables to be changed: A\$, E2, S1(1), T5(1,12), and Z1(1,12) were then entered from the keyboard, following which the command "RUN 1120" was entered. The program upon receipt of this command put these changed values into file 27.

HEADER INFORMATION FROM FILE 27

NUMBER OF CABLE POSITIONS (N1) =1
 NUMBER OF SEIS/CABLE (N2) =12
 IDENTIFICATION NUMBER ON TYPE OF SURVEY (N3) = 4
 A crosshole or hole-to-hole survey was run
 AREA (A\$) = [WATKINS]
 HOLE DESIGNATION (U\$) = 79-101A
 COUNTY AND STATE (C\$) = Adams County, Colorado
 DATE DATA TAKEN (D\$) = Feb. 14, 1980
 LATITUDE (L\$) = 39 48 45.6 AND LONGITUDE (O\$) = 103 23 12.3
 DETECTOR TYPE (R\$)=Piezoelectric AND SOURCE TYPE (S\$)=Shotgun
 ELEV OF TEST HOLE (E1) = 1650; ELEV OF SP HOLE (E2) = [1651.2]
 SP TO TEST HOLE DIST (O1) = 10

CABLE NO.	DEPTH OF SHOT	DEPTH UPPER DETECTOR	DEPTH LOWER DETECTOR
J	S1(1)	Z1(J,1)	Z1(J,12)
1	[98.2]	91.0	[102.1]

FOR CABLE POSITION 1:

SEIS NO.	ARRIVAL TIME	SEIS DEPTH
J, K	T5(1,K)	Z1(1,K)
1, 1	5.710	91.0
1, 2	5.190	92.0
1, 3	4.710	93.0
1, 4	4.820	94.0
1, 5	4.650	95.0
1, 6	5.050	96.0
1, 7	5.450	97.0
1, 8	5.580	98.0
1, 9	5.480	99.0
1, 10	5.390	100.0
1, 11	5.360	101.0
1, 12	[5.220]	[102.1]

Figure 46. Copy of screen display showing contents of file 27 after the changes of figure 45 were made.
 On this figure, changed values are enclosed in dashed-line boxes for emphasis.

Listing of program to review and change data file

```

100 PRINT "L_YOU HAVE SELECTED PROGRAM TO VIEW OR CHANGE ";
110 PRINT "CONTENTS OF A DATA FILE"
120 PRINT "G_G_G___NOTE: DATA TAPE MUST BE INSERTED WITHIN 4924"
130 INIT
140 DIM A$(20), C$(23), D$(12), G$(1), L$(10), O$(11), R$(16), S$(12), U$(7)
150 PRINT "___WHICH FILE DO YOU WISH TO EXAMINE? ";
160 INPUT F1
170 FIND @2:F1
180 PRINT "L_                                HEADER INFORMATION FROM FILE "; F1
190 READ @2:N1, N2, N3, A$, C$, D$, L$, O$, R$, S$, U$, E1, E2, O1
200 IF N3=2 THEN 240
210 PRINT "NUMBER OF CABLE POSITIONS (N1) = "; N1
220 PRINT "NUMBER OF SEIS/CABLE (N2) = "; N2
230 GO TO 250
240 PRINT "NUMBER OF DETECTOR POSITIONS WITHIN THE HOLE = "; N2
250 PRINT "IDENTIFICATION NUMBER ON TYPE OF SURVEY (N3) = "; N3
260 GO TO N3 OF 270, 290, 310, 340
270 PRINT "    A downhole survey using a velocity cable was run"
280 GO TO 350
290 PRINT "    A downhole survey with a single detector group was run"
300 GO TO 350
310 PRINT "    An inhole survey (source and detectors in same hole)";
320 PRINT "    was run"
330 GO TO 350
340 PRINT "    A crosshole or hole-to-hole survey was run"
350 PRINT "AREA (A$) = "; A$
360 PRINT "HOLE DESIGNATION (U$) = "; U$
370 PRINT "COUNTY AND STATE (C$) = "; C$
380 PRINT "DATE DATA TAKEN (D$) = "; D$
390 PRINT "LATITUDE (L$) = "; L$; " AND LONGITUDE (O$) = "; O$
400 PRINT "DETECTOR TYPE (R$) = "; R$; " AND SOURCE TYPE (S$) = "; S$
410 IF N3<>3 THEN 460
420 PRINT "ELEV OF TEST HOLE (E1) = "; E1; " ELEV OF";
430 PRINT "CHECK SEIS (E2) = "; E2
440 PRINT "CHECK SEIS OFFSET = "; O1
450 GO TO 530
460 PRINT "ELEV OF TEST HOLE (E1) = "; E1; " ELEV OF SP HOLE (E2) = "; E2
470 PRINT "SP TO TEST HOLE DIST (O1) = "; O1
480 IF N3=2 THEN 500
490 GO TO 530
500 DIM R1(N2), S1(N2), T1(N2), T5(1, N2), Z1(1, N2)
510 N1=N2
520 GO TO 540
530 DIM R1(N1), S1(N1), T1(N1), T5(N1, N2), Z1(N1, N2)
540 FIND @2:F1
550 READ @2:N1, N2, N3, A$, C$, D$, L$, O$, R$, S$, U$, E1, E2, O1, R1, S1, T1, T5, Z1
560 GO TO N3 OF 760, 570, 610, 680
570 N1=N2
580 PRINT "___SEIS. NO. REFERENCE TIME    SHOT DEPTH    UP";
590 PRINT "HOLE TIME"
600 GO TO 790
610 PRINT "___CABLE NO.    CHECK SHOT TIME    SHOT DEPTH"
620 IMAGE 2X, 2D, 12X, 3D, 2D, 12X, 3D, D
630 PRINT "    J                T5(J, 1)                S1(J)"
640 FOR J=1 TO N1
650 PRINT USING 620: J, T5(J, 1), S1(J)
660 NEXT J
670 GO TO 830
680 PRINT "___CABLE    DEPTH OF    DEPTH UPPER    DEPTH LOWER"
690 PRINT "    NO.    SHOT    DETECTOR    DETECTOR"
700 IMAGE X, 2D, 7X, 3D, D, 9X, 3D, D, 12X, 3D, D
710 PRINT "    J                S1(1)                Z1(J, 1)                Z1(J, 12)"
720 FOR J=1 TO N1
730 PRINT USING 700: J, S1(J), Z1(J, 1), Z1(J, 12)
740 NEXT J
750 GO TO 830

```



```

760 PRINT "__CABLE NO. REFERENCE TIME SHOT DEPTH UP";
770 PRINT "HOLE TIME"
780 IMAGE 2X, 2D, 9X, 3D, 2D, 9X, 3D, D, 9X, 3D, D
790 PRINT " J R1(J) S1(J) T1(J)"
800 FOR J=1 TO N1
810 PRINT USING 780: J, R1(J), S1(J), T1(J)
820 NEXT J
830 PRINT
840 J=1
850 IF N3=2 THEN 890
860 FOR J=1 TO N1
870 IF N3=2 THEN 890
880 PRINT " FOR CABLE POSITION "; J; ":"
890 PRINT "SEIS NO. "; " ARRIVAL TIME"; " SEIS DEPTH"
900 PRINT " J, K T5("; J; ", K) Z1("; J; ", K)"
910 FOR K=1 TO N2
920 IMAGE X, 2D, ", ", 2D, 6X, 4D, 3D, 8X, 3D, D
930 PRINT USING 920: J, K, T5(J, K), Z1(J, K)
940 NEXT K
950 IF N3=2 THEN 980
960 NEXT J
970 GO TO 990
980 N1=1
990 PRINT "__DO YOU WANT TO CHANGE ANY OF THESE DATA? (Y OR N) ";
1000 INPUT G$
1010 IF G$="N" THEN 1140
1020 PRINT "__PROCEDURE TO MAKE CHANGES IS AS FOLLOWS:"
1030 PRINT " 1. Identify name of variable to be changed; for example, "
1040 PRINT " A$, T1(2), Z1(2,4)"
1050 PRINT " 2. Enter variable name and its new value; for example, "
1060 PRINT " A$="WATKINS", T1(2)=12.5, Z1(2,4)=30"
1070 PRINT " 3. After all changes have been made, enter RUN 1120"
1080 PRINT "G_G_G___ARE YOU READY TO PROCEED? (Y OR N) ";
1090 INPUT G$
1100 IF G$="N" THEN 1080
1110 END
1120 FIND @2:F1
1130 WRITE @2:N1, N2, N3, A$, C$, D$, L$, O$, R$, S$, U$, E1, E2, O1, R1, S1, T1, T5, Z1
1140 PRINT "G_G_G___PROGRAM COMPLETED"
1150 END

```